
INTRODUCCIÓN

¿Que es Fivewin?

FiveWin es una librería para CA-Clipper que permite realizar programas bajo Windows, sin embargo dentro del paquete hay mucho más de lo que se pueda ver a simple vista ya que FiveWin proporciona a CA-Clipper un auténtico entorno orientado al objeto con creación de nuevas clases, herencia, polimorfismo, etc.

FiveWin utiliza esta potencia OOPS para crear sus propias clases y desde ellas acceder a todo el API de Windows sin que el programador tenga necesidad de entenderse con él mismo, sin embargo y debido a que FiveWin incorpora en el paquete todos los fuentes de dichas clases es realmente sencillo modificar o subclasificar cualquiera de ellas para acomodarla a nuestras necesidades. Además el uso de todas y cada una de las clases se puede realizar utilizando comandos al más puro estilo Dbase, por lo que la utilización de FiveWin es realmente sencilla.

FiveWin le da acceso a casi la totalidad de recursos de Windows como por ejemplo:


( Entorno MDI (Multiple Document Interface)


( Menús flotantes


( Barras de Botones


( Bitmaps (Imágenes)


( Fuentes True Type


( Cursores


( Iconos


( Cajas de diálogo


( Acceso a DLL


( Controles VBX


( Clipboard


( Enlaces DDE


( Ficheros INI


( Metafiles


( Meters


( Brushes


( Sonido (Archivos WAV, MID)


( Vídeo (AVI)


( ODBC

Además FiveWin incorpora la clase TWBrowse para poder realizar browse bajo Windows y la clase TReport para realizar cualquier informe.

Compilar

Para compilar una aplicación realizada con FiveWin compile como lo hacia normalmente:


CLIPPER  Test  /n/w

Casi todas las aplicaciones escritas con Fivewin necesitan que se incluya la siguiente línea:


 #include "FiveWin.ch"

Linkar

FiveWin consta de las siguientes librerías:

· FiveWin.LIB

· WinApi.LIB

· 501_520


La librería 501_502 sólo la necesita sí va a utilizar Microsoft LINK.

Necesita un linkador para aplicaciones Windows, el cual no es proporcionado por CA-Clipper por lo que deberá adquirirlo a parte. Puede utilizar el linkador de Microsoft

LINK.EXE versión 5.3 o posterior, o Blinker 3.0 o posterior siendo éste último el más recomendable.

Sintaxis para Microsoft LINK:
 
Link //se:700 //NOE <SuAplicacion>,,nul, FiveWin WinApi  ;




Clipper501_520,..\include\FiveWin.def

Sintaxis para Blinker 3.0:

Blinker fi SuAplicacion @FiveWin.lnk

El fichero Fivewin.lnk viene incorporado con FiveWin.

Arquitectura

La arquitectura de Fivewin esta basada en los siguientes niveles:

1 Comandos xBase

2 Clases y objetos

3 Funciones

4 Windows API

Utilizando los comandos xBase podremos rápidamente a producir nuestros primeros programas. Cuando necesite un poco más de control tendrá que descender un poco más, tenga en cuenta que cuanto más descienda más control posee pero sin embargo disminuye su productividad.

FiveWin hace un uso extensivo de nuevos objetos, sin embargo y gracias al preprocesador de CA-Clipper el uso de los mimos esta disfrazado en forma de comandos cuya sintaxis es más natural, sin embargo es posible sí se desea acceder directamente a las clases sin necesidad de utilizar los comandos xBase.

Las Clases que incorpora FiveWin para el desarrollo de aplicaciones Windows hace que no sea necesario en casi ningún caso el tener que profundizar hasta el último nivel del API de Windows, sin embargo sí lo desea puede acceder a él sin problemas.

Utilidades necesarias

A parte de FiveWin y CA-Clipper le serán necesarias una serie de utilidades:

· linkador para Windows

· compilador de recursos

· Generador visual de recursos (Resource Workshop de Borland)

El linkador de recursos es necesario ya que el suministrado por CA-Clipper no es capaz de realizar ejecutables para Windows.

El compilador de recursos le permitirá incluir en su ejecutable final recursos generados a través del Resource Workshop, por ejemplo, son recursos las imágenes, los iconos que se encuentran en casi todos los programas para Windows.

También es recomendable pero no imprescindible la obtención de un compilador de ayudas por sí desea realizar ayudas bajo Windows y documentación sobre el API de Windows en formato Help de Microsoft para su consulta cuando queramos acceder al API.

Por ejemplo el icono que poseen todas las aplicaciones Windows cuando se iconizan es un recurso que normalmente se guarda en un fichero con extensión ICO pero que como veremos en posteriores capítulos se puede encontrar en un fichero de extensión RC o incluso dentro de una DLL. Para adjuntar dicho icono al ejecutable habrá que crear un fichero RC con el Resource WorkShop y a continuación utilizar la herramienta RC.EXE para incluirlo definitivamente en el ejecutable.

Como organizar su trabajo con Windows

Aunque los programas que va a realizar con FiveWin van a ser auténticos ejecutables tipo Windows, sin embargo la mayoría de las herramientas que va a utilizar son tipo DOS, como el compilador de CA-Clipper, el linkador Blinker 3.0 y casi seguramente su editor de programación favorito.

No es necesario que empiece a adquirir ningún editor para Windows, y de hecho puede seguir utilizando el mismo de siempre, le recomendamos que organice su entorno de programación de la siguiente manera:

· Cree un grupo nuevo en el Administrador de programas en el cual incluirá todas las aplicaciones que este desarrollando.

· Ejecute su editor desde una ventana DOS para ello deberá cargar el editor a través de un archivo PIF indicando en el  mismo que va a ser ejecutado en una ventana DOS.

· Si no puede ejecutar el compilador y el linkador desde su editor cree un nuevo icono de programa en el administrador de programas para ejecutar el compilador y el linkador desde Windows.

En cualquier caso para desarrollar cómodamente desde Windows es prácticamente necesario un ordenador 486 con 4Mb de RAM.

Instalación de Fivewin

FiveWin se instala automáticamente en su ordenador bajo la siguiente estructura:

( C:\Clipper5

( Fwxxx
Donde xxx es la versión de FiveWin

( Bitmaps
algunas imágenes para incluir en sus propias aplicaciones

( Cursors
algunos cursores para incluir en sus propias aplicaciones 

( Dll
Una DLL vacía para utilizar como contenedor de sus recursos

( Icons
algunos iconos para incluir en sus propias aplicaciones

( Ide 
Entorno IDE de FiveWin (no operativo por el momento)

( Include 
Ficheros de Cabecera tanto para CA-Clipper como para C

( Lib 
Librerías de FiveWin 

( Manual 
Guía Norton

( Samples
Tutoriales y ejemplos

( Source
Fuentes

( Classes
Clases

( Function 
Funciones

( Tools 
Utilidades

( Winapi 
Conexión CA-Clipper y Windows API 

Por defecto Fivewin es instalado en el directorio de CA-Clipper , pero puede modificarlo sí lo desea.

· Es preferible que cada vez que instale FiveWin lo haga en un directorio diferente, y pruebe la nueva versión antes de destruir la que ya tiene.

· Lea la información contenida en el archivo Whatsnew.txt para conocer las últimas mejoras realizadas en FiveWin así como los posibles errores que estuviesen corregidos.

· Si usted posee la versión protegida por contraseña (como es el caso de la versión existente en España) necesitará registrar su versión de FiveWin, para ello deberá pedir a su distribuidor que le proporcione una clave de registro. Antes de registrarse cerciórese de que sólo tiene una versión de Fivewin.lib que es la que quiere registrar.

· Es posible que sea necesario modificar el archivo Samples\Build.bat para acomodarlo a los ‘paths’ y ‘sets’ de su ordenador.

· Si la instalación se ha realizado correctamente vaya al directorio Samples e intente crear su primera aplicación realizando la siguiente instrucción:

SAMPLES\Build Tutor01

· Se creará un programa de nombre Tutor01.Exe que funciona realmente en Windows. Vaya a Windows e intente ejecutarlo.

Compatibilidad con Fivewin

Cuando hablamos de compatibilidad hay que distinguir entre compatibilidad con FiveWin y compatibilidad con Windows, en principio sí una librería es compatible con Windows también es compatible con FiveWin, para que una librería sea compatible con Windows ha de poder correr en modo protegido y además no debe acceder al hardware directamente ni debe utilizar la pantalla ya que es dominio exclusivo de Windows y por lo tanto a de utilizar los servicios que Windows proporciona para acceder a dichos recursos.

Debido a que FiveWin incorpora su propio motor OOPS es incompatible con otros motores OOPS existentes como es Class(y) él cual es utilizado por algunas librerías comerciales, por lo que FiveWin es incompatible con Class(y) y con todas aquellas funciones de otras librerías que utilicen un run-time de Class(y), por ejemplo la librería Funcky II lo utiliza, sin embargo se puede usar perfectamente con FiveWin, siempre y cuando no se utilicen las clases que vienen incorporadas en la librería de Funcky II.

Son compatibles con FiveWin las librerías de Comix, Six 2.0 y parte de Funcky entre otras. En cuanto empiece a programar bajo Windows se dará cuenta que gran cantidad de las librerías que estaba usando ya no son necesarias porque el propio Windows le da soporte para realizar lo mismo , como por ejemplo librerías gráficas, de acceso a redes locales, envío de faxes, etc. Además descubrirá nuevas formas de librerías como son los controles VBX y las librerías DLL los cuales se pueden utilizar perfectamente desde FiveWin y sin ninguna limitación.

Ejercicios

1. Que flags son necesarios y cuales son recomendables para el uso del compilador de CA-Clipper con FiveWin?

2. Modificar Build.Bat para el siguiente entorno: CA-Clipper en directorio D:\CLP

3. Prepare su entorno de programación bajo Windows.

ELEMENTOS FUNDAMENTALES DE UNA APLICACIÓN DE GESTIÓN EN WINDOWS

Ventana Principal

Toda aplicación bajo Microsoft Windows suele tener una ventana principal, normalmente dicha ventana tendrá un título o ‘Caption’, un menú, una barra de botones y una barra de mensajes en su parte inferior, aunque como veremos más adelante no es necesario realmente la existencia de todos estos componentes.

Existen dos posibilidades desde un punto de vista genérico de crear la ventana principal, que son:

· Ventana SDI

· Ventana MDI

SDI significa Single document interface, es decir, un único documento o ventana para toda la aplicación y MDI que significa Multiple document interface, en el cual hay múltiples documentos o ventanas dentro de la ventana principal. Ejemplos de ventanas MDI es el propio administrador de programas o el administrador de archivos, en los cuales puede observar que existe una ventana principal y dentro de la misma existen otras ventanas ‘hijas’. En el caso del Administrador de Programas cada ventana hija corresponde con un grupo del administrador de programas.

El diseño MDI es más apropiado cuando el programa a realizar suponga la apertura o la utilización de múltiples instancias del mismo tipo de documentos, es decir, un programa para editar ficheros DBF genérico es preferible realizarlo con un diseño MDI, en el resto de los casos recomendamos que se utilice un diseño SDI el cual es más sencillo de mantener y de comprender por parte del usuario. Es importante que cuando cree una aplicación tenga en cuenta que su aplicación la va a utilizar un usuario que en la mayoría de los casos desconoce el interface de Windows y la complejidad de su aplicación en su globalidad, por lo que si le permite tener múltiples ventanas abiertas, cada una de ellas de distinto tipo, es posible que le cueste un gran trabajo aprender y utilizar su aplicación. 

Las ventanas puede tener varios estilos o propiedades, como por ejemplo que tenga título, que se puede mover, que se pueda redimensionar, etc.. FiveWin le permite indicar en la creación de la ventana el estilo que ha de tener.

Desde nuestra ventana principal podremos utilizando un menú abrir nuevas ventanas si lo deseamos, sin embargo lo habitual es abrir cajas de diálogo que aunque realmente son ventanas sin embargo poseen características especiales que las diferencian del resto de las ventanas, que son:

· Su ejecución es modal, es decir, no podrá volver a la ventana principal hasta que no cierre la caja de diálogo, por ejemplo pulsando un botón de ‘Aceptar’ que tuviese la caja de diálogo. Se pueden realizar también cajas de diálogo no modales, sin embargo no es lo habitual.

· Las cajas de diálogo pueden ser definidos en recursos externos.

· Los diseñadores visuales de recursos sólo manejas cajas de diálogo, no ventanas.

· Todos los ‘controles’ de Microsoft Windows se suelen colocar en cajas de diálogo y casi nunca en ventanas.

· Las cajas de diálogo no suelen ser redimensionables.

· Las cajas de diálogo no pueden llevar menús.

En el capítulo de cajas de diálogo se verá con más detalle las características de las mismas.

La forma de crear una ventana bajo FiveWin es la siguiente:

#include “Fivewin.ch”

Static oWnd

Function Main()


DEFINE WINDOW oWnd TITLE "Mi programa"


ACTIVATE WINDOW oWnd MAXIMIZED

return nil

En este ejemplo esta generando la ventana más típica con coordenadas y estilo prefijados, pero cuando la active se abre en modo maximizado lo que quiere decir que ocupara toda la ventana.

Nota:
Para más información sobre la diferencia entre ventana en estado normal o maximizado consulte el tutorial de Microsoft Windows.

Si desea colocar la ventana en una posición en concreto:

... FROM nArriba, nIzquierda TO nAbajo, nDerecha ...

De esta forma simulamos las coordenadas tipo texto que utilizamos normalmente con CA-Clipper de fila y columna, ya que Microsoft Windows utiliza coordenadas en pixeles. Si desea utilizar coordenadas en pixeles como lo hace Microsoft Windows simplemente introduzca la cláusula PIXELES en el comando.

A la ventana se le pueden aplicar una gran variedad de estilos, por defecto la ventana posee el estilo definido en el Microsoft Windows API como OVERLAPPED-WINDOW que significa que posee las siguientes características:

· Caption o Título

· Menú de sistema

· Redimensionable

· Botones de minimizar y maximizar

Si desea especificar otros estilos FiveWin le ofrece las siguientes cláusulas:

· NOSYSMENU

· NOCAPTION

· NOICONIZE

· NOZOOM

También puede especificar cualquier combinación de ellas utilizando directamente los flags del API de Microsoft Windows y utilizando la función nOr() que incluye FiveWin:

... STYLE nOr( WS_BORDER, WS_MAXIMIZE, ... )

Los diferentes estilos posibles los encontrará en el fichero Include\Winapi.ch que es automáticamente cargado por FiveWin cuando se carga Fivewin.ch o también en la ayuda on-line de Microsoft Windows API que puede encontrar en cantidad de productos comerciales de programación bajo Windows

Si desea que su ventana principal sea una ventana MDI simplemente especifique la cláusula al crear la ventana:

DEFINE WINDOW oWnd TITLE "A MDI environment" MDI

FiveWin controla automáticamente todos los mensajes que recibe la ventana principal y usted sólo debe procesar aquellos que le interese utilizando un sencillo lenguaje de comandos que siempre empiezan por la cláusula ‘ON’ los cuales se asignan cuando se activa la ventana.

ACTIVATE WINDOW oWnd ;


ON PAINT ... ;


ON LEFT CLICK ... ;


ON RESIZE ... ;

Cada una de las cláusulas que podemos controlar es lo que llamaremos un ‘Evento’, es decir, cuando se produzca el evento de pulsación del botón izquierdo del ratón y éste se encuentre dentro de nuestra ventana se ejecutará el código que hayamos especificado en la cláusula ‘ON LEFT CLICK’. 

Al principio no necesitará manejar casi ningún evento, pero cuando tenga más experiencia con FiveWin es posible que le sean de gran utilidad.

También puede si lo desea incluir barras de desplazamiento a la ventana para ello sólo debe incluir la cláusula VSCROLL para barras verticales y HSCROLL para las horizontales.

DEFINE WINDOW oWnd ... VSCROLL

o

DEFINE WINDOW oWnd ... HSCROLL

o las dos

DEFINE WINDOW oWnd ... VSCROLL HSCROLL

Si coloca barras de desplazamiento también le será posible interceptar los mensajes o eventos que ellas reciben de la misma forma que el resto de las cláusulas ‘ON ...’

ACTIVATE WINDOW oWnd ...

         ON UP ...

         ON DOWN ...

         ON LEFT ...

         ON RIGHT ...

         ON PAGEUP ...

         ON PAGEDOWN ...

Nuestro objeto ventana oWnd tendrá a su vez definidos dos objetos Scrollbar que serán:

oWnd:oVScroll y oWnd:oHScroll

Por lo que podrá manejar directamente los eventos que recibe cada uno de esos objetos.

Para finalizar la ejecución de la ventana utilice el método ‘End’.

oWnd:End()

Esta instrucción indica al objeto ventana que debe finalizar. Si ha especificado la cláusula VALID cuando activo la ventana, entonces la instrucción que hubiese indicado será evaluada y si retorna .T. la ventana terminará pero no si retorna .F. . Por ejemplo:

ACTIVATE WINDOW oWnd ;

          VALID MsgYesNo( "¿Desea Terminar?" )

Cuando intente cerrar la ventana desde el menú de sistema o pulsando Alt+F4 el programa le mostrará una caja de diálogo en la cual habrá dos botones “Si” y “No” con el mensaje “¿Desea terminar?”, si pulsa “Si” la función devolverá .T. y se cerrará la ventana, por el contrario si pulsa “N” la función devolverá .F. y la ventana no se cerrará.

Menús

La típica estructura de un programa de Microsoft Windows es crear una ventana principal y asignar un menú a la misma desde el cual accederemos a todas las opciones de la aplicación.

Desde el menú el usuario puede acceder a todos los procesos del programa.

Debe intentar realizar previamente un diseño de su menú sin hacerlo demasiado complicado, lo más sencillo posible y creando submenús para englobar opciones del mismo tipo.

Con FiveWin no es necesario la típica secuencia ‘DO CASE’ para controlar que opción ha sido elegida, cuando un elemento del menú es definido se especifica directamente la acción que se va a producir cuando se elija.

MENU oMenu


MENUITEM <cItem> MESSAGE <cMessage> ACTION <uAction>
MENUITEM <cItem> MESSAGE <cMessage>  


MENU                                 



MENUITEM <cItem> MESSAGE <cMessage> ACTION <uAction>


ENDMENU


...


SEPARATOR


MENUITEM <cItem> MESSAGE <cMessage> ACTION <uAction>

ENDMENU

‘MESSAGE y ACTION’ son cláusulas opcionales. Si no se especifica ninguna acción entonces nada ocurrirá cuando elija dicho elemento del menú a no ser de que haya un submenú que cuelgue de dicha opción.

Se pueden especificar todos los submenús que desee y con el anidamiento que sea preciso, para crearlo sólo deberá anidar dentro de una estructura MENÚ ... ENDMENU una nueva estructura MENÚ ... ENDMENU 

MENUITEM <cItem> MESSAGE <cMessage>


MENU


  MENUITEM <cItem> MESSAGE <cMessage> ACTION <uAction>

   ...

 ENDMENU

MENUITEM <cItem> MESSAGE <cMessage> ACTION <uAction>

Para activar el menú deberá primero asignarlo a una ventana. La forma más sencilla de hacerlo es:

DEFINE WINDOW <oWnd> <otherClauses> ;

       MENU BuildMenu()

...

...

Function BuildMenu()


local oMenu

  MENU oMenu

    MENUITEM <cItem> MESSAGE <cMessage> ...

    ...

  ENDMENU

return oMenu

 o  lo puede realizar cuando la ventana este creada con el comando:

       SET MENU OF <oWnd> TO <oMenu>

Barra de botones

La barra de botones se ha convertido un un standard dentro de los componentes que podemos encontrar en casi todas las aplicaciones para Windows.

El propósito de dicha barra no es otro que ofrecer un acceso rápido a aquellas opciones más utilizadas de su aplicación. 

Una barra de botones bien diseñada ayuda enormemente en el aprendizaje y uso de cualquier aplicación. No utilice imágenes sin sentido, por el contrario intente utilizar imágenes que de una forma intuitiva indiquen la acción a realizar cuando sea pulsada con el ratón.

La barra de botones es una área de la ventana en la cual se van a alojar una serie de botones con imágenes en su interior. Las acciones que realizan dichos botones suelen ser opciones duplicadas ya existentes en opciones de menú debido a que a las barras de botones no se puede acceder desde el teclado.

Existe la posibilidad de seleccionar un aspecto 3D a la barra de botones simplemente añadiéndole la cláusula 3D en la creación de la misma.

Normalmente la barra de botones suele estar en la parte superior de la ventana por debajo del menú, sin embargo FiveWin tiene la habilidad de poder colocar la barra de botones donde se desee simplemente pulsando el botón derecho del ratón.

Todas las ventanas poseen un objeto :oBtnBar que es donde se guarda la referencia a la barra de botones.

Para incluir una barra de botones en cualquier ventana se utiliza el siguiente comando:

#xcommand DEFINE BUTTONBAR [ <oBar> ] ;

      [ <size: SIZE, BUTTONSIZE, SIZEBUTTON > <nWidth>,      
       <nHeight> ] ;

      [ <_3d: 3D, 3DLOOK> ] ;

      [ <mode: TOP, LEFT, RIGHT, BOTTOM, FLOAT> ] ;

      [ <wnd: OF, WINDOW, DIALOG> <oWnd> ] ;

      [ CURSOR <oCursor> ] ;

Barra de mensajes:

La barra de mensajes se ha convertido también en un standard que podemos encontrar en multitud de aplicaciones para Microsoft Windows .

Una barra de mensajes es un control que se coloca en la parte inferior de la ventana que muestra información relacionada con el elemento de menú que este actualmente seleccionado o en el caso de una barra de botones con el botón sobre el que se encuentre el cursor del ratón.

FiveWin muestra la barra de botones con efecto 3D sin necesidad de tener que especificarlo y opcionalmente le permite colocar dichos mensajes centrados en la barra de mensajes.

Actualmente en fase Beta y probablemente disponible cuando este leyendo este documento la barra de botones implementará la posibilidad de tener múltiples columnas con información de tiempo, fecha, estado del teclado, etc.

Cajas de Diálogo

En Microsoft Windows la introducción de cualquier dato por parte del usuario en una aplicación se hace a través de las cajas de diálogo. Las cajas de diálogo son ventanas especializadas en recogida de información a través de unos controles standard de Microsoft Windows.

No es practica recomendable utilizar la ventana principal de una aplicación para recoger datos.

FiveWin soporta dos métodos de utilización de cajas de diálogo:

· Construidas con código fuente

· Desde recursos generados con un editor de recursos para Microsoft Windows

El segundo método es el más aconsejable ya que le permite definir las cajas visualmente de una forma muy rápida e intuitiva, para utilizar este método necesitará un editor de recursos como el Resource WorkShop de Borland, el AppStudio de Microsoft o el Resource Studio de Symantec.

En Microsoft Windows las cajas de diálogo pueden ser de dos categorías:

· Modales

· No modales

Las modales son aquellas que no permiten al usuario realizar ninguna acción hasta que se cierre dicha caja de diálogo, en MSDOS siempre se han utilizado cajas de diálogo modales. Las cajas de diálogo no modales permiten hacer al usuario otras cosas mientras están activas. En principio y hasta que este un poco más familiarizado con Microsoft Windows utilice únicamente cajas modales ya que hay que saber muy bien lo que se esta haciendo cuando se utilizan cajas no modales.

Una caja de diálogo es un objeto y para crearla utilizará la típica sintaxis también utilizada en la creación de ventanas pero con algunas variaciones:

DEFINE DIALOG oDlg ;

 [ FROM <nTop>, <nLeft> TO <nBottom>, <nRight> ] ;

 [ NAME <cName> ] ;

La cláusula ‘NAME’ indica el nombre de dicho diálogo si se hubiera creado a través de un editor de recursos.

Después deberá asignar todos los controles que va a tener dicha caja de diálogo si la esta diseñando directamente desde código fuente, por el contrario si la ha creado con un editor de recursos los controles ya estarán creados y sólo necesitará redefinirlos para modificar alguna característica de los mismos si así lo desea.

Si esta creando la caja de diálogo directamente la sintaxis es muy parecida a la de los comandos de CA-Clipper.

@ <nRow>, <nCol> ... OF oDlg

Recuerde especificar siempre ‘OF oDlg’ de esta forma el control sabrá a quien le pertenece (su contenedor).

Si esta creando la caja de diálogo desde recursos es posible que necesite redefinir algún control, para ello deberá utilizar el comando 

REDEFINE ID ... OF oDlg

que será explicado en profundidad más adelante.

Para activar la caja de diálogo utilice el comando ‘Activate’

ACTIVATE DIALOG oDlg

Nuestro primer programa con FiveWin

Vamos a intentar realizar nuestro primer programa con Fivewin el cual va a tener las siguientes características:

· Ventana principal tipo SDI

· Menú con una única opción de cerrado de la ventana

· Barra de mensajes mostrando nuestro nombre

· el típico mensaje en el centro de la ventana de “Mi primer programa con FiveWin”

· Validación de abandono de la aplicación mediante la caja de diálogo predefinida ‘MsgYesNo’

El código fuente de este pequeño programa lo puede encontrar en el disquete suministrado en el curso con el nombre:

tutor2\ejemplo1.prg.

Se reproduce aquí por comodidad.

#include “FiveWin.ch”

STATIC oWnd

FUNCTION Main()


DEFINE WINDOW oWnd TITLE “Mi primer programa” ;


  MENU BuildMenu()

   SET MESSAGE OF oWnd TO “Mi nombre” CENTERED

   ACTIVATE WINDOW oWnd VALID MsgYesNO(“¿Desea salir?”)

return nil

//-----------------------------------------------------//

Function BuildMenu()


local oMenu


MENU oMenu

      MENUITEM “Fin” ACTION oWnd:End()


ENDMENU

Return oMenu

Ejercicios

1. Intente añadir al programa de ejemplo una barra de botones estilo  3D.

2. Realice un programa con una opción de menú en su ventana principal que al ejecutarse abra una caja de diálogo con el título “Mi primer caja de diálogo”.

3. Realice un programa en el cual cada vez que se pulse el botón izquierdo del ratón suene un pitido.

CONTROLES DE WINDOWS

Que son los controles

Los controles son realmente ventanas con características especiales para facilitar la entrada de datos y la comunicación de la aplicación con el usuario. Decimos que son ventanas ya que todos ellos son objetos heredados de la clase Ventana (tWindow) pero sin embargo a la vista se parecen muy poco, pero lo importante ha tener en cuenta es que son objetos y que muchas de sus características son heredades de la clase genérica ventana.

Los controles suelen incorporarse sólo en las cajas de diálogo  aunque también pueden incluirse en las ventanas si se desea. CA-Clipper (bajo MSDOS) sólo posee dos tipos de controles que serían el control GET y el control SAY que en Microsoft Windows se llaman EDIT TEXT y TEXT. Sin embargo Microsoft Windows incluye más controles que mejoran substancialmente el interface con el usuario.

Todos los controles heredan sus propiedades de la clase tControl que a su vez es una clase heredada de tWindow. Cada control de Microsoft Windows esta implementado en FiveWin con una nueva clase para hacer que su uso sea realmente sencillo. Usted no tiene porque saber en principio como es la jerarquía de clases de FiveWin sin embargo es recomendable que desde el principio vaya adquiriendo la sensación de comprender como realmente es FiveWin por dentro, ya que ello le permitirá explotarlo al máximo.

Los controles deben estar contenidos  siempre en una ventana o caja de diálogo y no pueden tener ningún menú asociado. Todos los controles tienen un número identificador del mismo por el cual podemos acceder cuando utilicemos controles creados a través de un editor de recursos.

En este capítulo se hace una relación de los controles standard y los más importantes que se pueden utilizar con FiveWin indicando las características de cada uno de ellos, su funcionalidad y algunos consejos para su utilización, sin embargo no se da una explicación exhaustiva sobre las distintas cláusulas de creación de cada uno de ellos con comandos xBase, ni tampoco una relación de todos los métodos y datos de los mismos. Para obtener mayor información sobre cada uno de estos controles utilice la guía Norton que acompaña a la librería FiveWin.

En una caja de diálogo donde se encuentren varios controles solamente uno de ellos recibe información del usuario, se dice que ese control tiene ‘Focus’, cuando un control tiene focus es fácilmente reconocible, en el caso de un Get podrá observar el cursor del teclado y en el caso de un botón observará un cuadrado fino alrededor del título del mismo y un marco más oscuro que en el resto de los botones. No todos los controles pueden tener focus por carecer de sentido, es el caso de un Say o Text.

Para acceder a un control sólo hay que posicionar el ratón encima de él y hacer un click del botón izquierdo, con el teclado hay dos formas de acceder, la primera consiste en pulsar la tecla ‘TAB’ con lo que se modifica el focus al siguiente control hasta que lleguemos al control deseado, pulsando ‘TAB’ avanzamos hacia el siguiente control y con ‘May+TAB’ retrocedemos al anterior control; el segundo método consiste en pulsar el acelerador si lo tuviese de dicho control.

Un acelerador es una pulsación de teclado que directamente nos lleva al control deseado, para crear un acelerador hay que poner el título o caption del control el signo ‘&’ delante de la letra que queremos actúe de acelerador y pulsando la tecla ‘Alt’ más la letra indicada saltaremos directamente al control. En el caso de los controles botones cuando pulsemos el acelerador se producirá la pulsación del mismo. Ha de tener la precaución de no poner el mismo acelerador a dos controles ya que siempre saltará al primero de ellos. Hay botones que pueden recibir focus pero sin embargo no tienen ningún título o caption, por lo que en principio no es posible establecer ningún acelerador para los mismos, sin embargo si es posible el truco consiste en colocar el acelerador en el control SAY que suelen tener bien encima a un lado e indicar que dicho control SAY tiene la propiedad de grupo, en el próximo capítulo cuando se estudie el Editor de Recursos se podrá observar con mayor claridad el funcionamiento.

Aquellos cursores que pueden tener varios estados, como por ejemplo un botón: pulsado o no pulsado, un checbox: verdadero o falso, etc. se puede cambiar su estado pulsando dicho control con el botón izquierdo del ratón o bien pulsando la tecla espaciadora cuando dicho control tenga focus.

Un control puede estar habilitado (Enabled) o deshabilitado (Disabled), en el primer caso podremos dar focus a ese control bien con el ratón o pulsando el tabulador sucesivamente hasta llegar a el mismo y en el segundo caso el control se le mostrará normalmente en un color gris para indicarle que no esta disponible.

En este capítulo encontrará muchas alusiones a la posibilidad de crear controles desde “recursos” que posiblemente no acierte a comprender como funcionan realmente, no se preocupe por que en el próximo capítulo serán estudiados en profundidad, simplemente tenga en cuenta que un recurso es un fichero externo donde se guarda parte del programa ejecutable que son los datos, así por ejemplo la definición de una caja diálogo con todos los controles que incorporé pueden ser definidos en un recurso y nuestro fichero ejecutable no tiene por que realizar la creación de los mismos, la idea fundamental que debe asimilar es que cualquier control puede ser creado directamente desde programa o bien redefinido desde un recurso externo creado con un editor de recursos.

Los controles pueden ser creados directamente o bien desde recursos,  en el primer caso deberá crear el control utilizando comandos estilo xBase e indicando las características principales del mismo como son coordenadas, texto si tuviese, etc.

@ <nRow>, <nCol> GET oGet VAR uVAr ;



OF oDlg ;



PICTURE “....” ;



VALID .... ;



WHEN .... ;



.... 

Sin embargo si ha creado el control con recursos (se explicará la forma de hacerlo en el siguiente capítulo) es posible que no necesite hacer nada con el control como puede ser por ejemplo un control SAY con el  cual no hay mucho más que hace que presentarlo dentro de la caja de diálogo, pero puede ser que necesite cambiar el comportamiento de dicho control para acomodarlo a sus necesidades, por ejemplo si es un campo Get debemos de tener alguna forma de saber que valor tiene dicho campo una vez que ha terminado la edición del mismo, o utilizar una máscara determinada de edición, para todos esos casos se utiliza el comando ‘REDEFINE’ que existe para todos los controles de FiveWin. Con este comando puede no sólo obtener una referencia al control diseñado vía recursos sino que además podrá modificar su comportamiento y apariencia. Como se dijo antes todo control tiene un identificador y gracias a él podemos acceder al mismo, la sintaxis es la siguiente:

REDEFINE CONTROL oControl ID .... OF oDlg

Por ejemplo para modificar la máscara de un Get y además guardar el resultado de su edición en la variable cData haríamos lo siguiente:

REDEFINE GET oGet VAR cData ID 101 PICTURE “!!!!!!” OF oDlg

Recuerde bien lo que acaba de leer por que esta es una de las bases de FiveWin y una de sus más potentes características. A diferencia de otros lenguajes que constantemente subclasifican cualquier clase en cuanto necesitan cambiar cualquier dato o comportamiento de la misma FiveWin utilizar la técnica de ‘redefinir’ que creemos que en muchos casos es más acertada y rápida. 

Es importante que cuando cree los controles indique siempre a que diálogo o ventana pertenecen con la cláusula OF:

... CONTROL ... OF oDlg

Cuando cree controles directamente desde código fuente deberá indicar la fila y columna donde desea que aparezcan, al igual que las ventanas o los diálogos, dicha fila y columna puede ser bien el equivalente a unas coordenadas tipo texto, o bien puede introducir las coordenadas directamente en pixeles incluyendo la cláusula PIXELES cuando cree el control. Si utilice coordenadas tipo texto tenga en cuenta que es posible introducir valores decimales en la fila y la columna. 

Es posible que el dato representado por un determinado control varíe durante la edición bien por otro control o por ejemplo por ser el campo de una base de datos y se ha movido el puntero de navegación del mismo, en estos caso es necesario refrescar la información contenida en dichos controles y para ello se utiliza un método genérico de FiveWin implementado en la clase base tWindow que es ‘refresh’.
oGet:Refresh()

Una técnica poco usada por muchos programadores de FiveWin pero muy cómoda es utilizar la cláusula UPDATE cuando se cree el control, cuando incluimos dicha cláusula podemos refrescar cuando lo deseemos todos los controles con dicha cláusula simplemente mandando el método Update a su diálogo contenedor.

oDlg:Update()

Es posible controlar la dinámica de la introducción de datos a través de los eventos que se producen en cada control, por ejemplo, cuando un control recupera focus usted puede indicar la ejecución de cierta instrucción simplemente inicializando el dato ‘bGotFocus’ que es un code-block de dicho control, ‘bGotFocus’ es un DATO de la clase tControl que si es distinto de NIL se evalúa cuando el control recibe focus, por ejemplo si queremos que suene un pitido cuando un control recibe focus haríamos lo siguiente:

oControl:bGotFocus := {|| tone(100,2)}

Al igual que el dato ‘bGotFocus’ existen otros datos de la clase para manejar el resto de los eventos que se pueden producir. FiveWin inicializa por usted los code-blocks más usuales dependiendo del tipo de control, por ejemplo si en un check box quiere controlar cuando cambia de estado existe una cláusula en la creación del mismo que es ‘ON CLICK’, realmente lo que hace FiveWin vía preprocesador es asignar al dato ‘bClicked’ de la clase tCheckbox la instrucción que usted ha introducido en el comando.

@ nRow, nCol CHECKBOX oChk VAR lData ;



    OF oDlg ;



    ON CLICK MiFuncion()

Si no introduce la cláusula ON CLICK podría hacer lo siguiente:

oChk:bClicked := {|| MiFuncion()}

y el resultado sería el mismo.

Tipos de controles

Microsoft Windows implementa de manera standard desde la versión 3.0 los siguientes controles:

Botones

Los botones de Microsoft Windows nos permiten establecer determinadas operaciones a realizar cuando estos sean pulsados, su funcionalidad es muy parecida a las opciones de menú en el sentido de que tienen una cláusula ‘ACTION’ en la cual se indica la instrucción a ejecutar cuando se pulse el botón.

Un botón puede llevar la cláusula DEFAULT lo cual indica que la acción de dicho botón se ejecutará cuando se pulse la tecla Intro aunque dicho control no tenga focus en ese momento, lógicamente sólo debe haber un botón con dicha cláusula en una misma caja de diálogo.

Si el botón es creado desde código fuente es posible  indicarle una imagen o fichero bitmap, esto es una característica de FiveWin que realmente no es standard de Microsoft Windows.

Barras de desplazamiento

Las barras de desplazamiento o ‘scrollbars’ son controles que pueden ser usados en infinidad de ocasiones, por ejemplo pueden servirnos para navegar por un fichero de texto, una base de datos o una imagen bitmap que no quepa en su totalidad en una caja de diálogo.

La característica principal de las barras de desplazamiento son las instrucciones que se pueden ejecutar al realizar determinadas operaciones con las mismas (eventos), estas son las principales:

· On up

· On down

· On page up

· On page down

· On Thumbpos

Los eventos ‘Up’ y ‘Down’ se producen cuando se pulsa con el ratón los botones superior e inferior de la barra de botones. ‘Page up’ y ‘Page down’ se producen cuando se pulsa con el ratón en el interior de la barra de botones bien por encima del botón-indicador o por debajo, el evento ‘Thumbpos’ se produce cuando realizamos un ‘drag’ del botón indicador.

Cuando cree una barra de desplazamiento deberá indicar valores máximos y mínimos para la misma, así como el incremento que se producirá en el valor cuando se produzca un evento tipo ‘Page up’ o ‘Page down’.

Las barras de desplazamiento pueden ser horizontales o verticales.

List boxes

Un listbox es un control que maneja listas que lógicamente están implementados bajo FiveWin con matrices.

Un listbox tiene asociado una matriz con todos los elementos de que consta y siempre hay un elemento que es el activo a no ser que la matriz este vacía.

Una vez creado el listbox se pueden añadir, modificar, borrar o elementos del mismo utilizando los métodos que FiveWin proporciona para el manejo de esta clase.

En la última versión de FiveWin se les ha incorporado la posibilidad de incluir algún bitmap o imagen a cada uno de los elementos.

Group boxes

Un Group box es un control que simplemente engloba a otros controles bien para que su presentación sea más intuitiva, o bien por que los controles a los que engloba actúan como un único control, este es el caso del los Radio buttons que son controles que muestran dos estados: ON|OFF, sin embargo cuando varios radio buttons están contenidos en un group box solamente uno de ellos puede estar ON. simulando efectivamente los botones de las antiguas radios.

Gets (Edit text)

Un Edit text es el equivalente a un Get de CA-Clipper. Debido a que el standard Edit text de Microsoft Windows no maneja ningún tipo de máscara o picture al estilo CA-Clipper, FiveWin implementa los gets utilizando un objeto Get genuino de CA-Clipper incluido como dato dentro del objeto tGet, a efectos prácticos esto significa que sí alguna vez ha manipulado directamente la clase Get original de CA-Clipper puede seguir haciéndolo con la simple precaución que deberá refrescar el control para que las modificaciones sean visualizadas.

La edición de un campo memo se realiza a través de un Get multi-line ya que para Microsoft Windows el control es el mismo a diferencia de CA-Clipper que no es capaz de mostrar un campo memo como un get más. Dicho get multi-line puede tener barras de scroll tanto verticales como horizontales.

Uno de los logros de FiveWin fue la implementación de la típica cláusula VALID y WHEN de CA-Clipper ya que de hecho estas dos cláusulas no están implementadas en Microsoft Windows. 

A diferencia de CA-Clipper se pueden definir la justificación del texto: izquierda, centrado o derecha.

Says (Text)

El control Say o Text es un texto fijo que puede ser o no multi-linea y que puede estar justificado a la izquierda, centrado o derecha.

Su funcionalidad es muy parecida al Say de CA-Clipper con la diferencia de que al tratarse de un objeto ahora puede ser manipulado cambiando su texto, etc.

Boxes

Los boxes son rectángulos que permiten mejorar la presentación de nuestras cajas de diálogo dando efecto 3D cuando se utilice dicho efecto con el comando al inicio del programa

SET 3D LOOK ON

Para que dicho efecto sea efectivo debe existir la librería Ctrl3d.dll a ser posible en el directorio Windows\System de su ordenador.

Radio buttons

Los Radio buttons son controles que muestran dos estados: ON|OFF, sin embargo cuando varios radio buttons están contenidos en un group box solamente uno de ellos puede estar ON. simulando efectivamente los botones de las antiguas radios. 

Se utilizan por ejemplo cuando desea un input de un número comprendido entre varios valores, por ejemplo, sí utiliza la variable nTarjeta para almacenar la posible tarjeta a utilizar, 1 podría ser Visa, 2 American Express y 3 Dinners, cada una de las opciones estaría representada por un radio button pero lógicamente cuando se seleccione uno de ellos el resto deberá estar en OFF.

Realmente son tres controles que actúan como uno sólo por este motivo quizás sea el control más complicado de utilizar. Cuando lo defina vía recursos no debe olvidar de incluirlos en un group box. Si lo define desde código fuente entonces se comporta como un único control y de la misma forma cuando establece el ‘REDEFINE’ para FiveWin pasan a ser un único control.

La creación de un radiobutton desde código fuente sería:

@ nRow, nCol RADIO oRadTarj VAR nTarjeta ;

             PROMPT “Visa”,”American Express”,”Dinners” ;

             OF oDlg

El ‘redefine’ de un radiobutton desde recursos sería:

REDEFINE RADIO oRadTarj VAR nTarjeta ;



ID 101,102,103 ;

         OF oDlg

Check boxes

Un check box es un control que puede tener dos estados: verdadero o falso, y por lo tanto se utiliza una variable lógica para recoger su valor. 

Es muy recomendable su uso en vez de los típicos Gets que se hacían con CA-Clipper de “Si”/”No”.

Combo boxes

Un combo box es un listbox desplegable, es decir, es un listbox que cuando no tiene focus nos muestra el elemento activo, pero cuando recibe focus se despliega como sí se tratase de un menú de persiana.

La ventaja fundamental de este tipo  de control es necesita mucho menos espacio que el listbox, pero sin embargo sólo podemos todas las opciones posibles cuando accedemos al mismo.

Su operativa es exactamente igual al listbox.



Controles Avanzados

Los controles avanzados son todos aquellos que no son standard de Microsoft Windows.

FiveWin implementa directamente algunos controles que no existen en Microsoft Windows como puede ser el control tWbrowse que permite manejar bases de datos, o el control tMeter que se utiliza para mostrar barras de progreso de por ejemplo una reindexación.

Borland a través de su editor de Recursos permite también utilizar los ya famosos BWCC (Borland Window  Common Control) para ello hay que utilizar en nuestros programas la librería de Borland BWCC.DLL.

FiveWin le permite también utilizar controles creados por otros fabricantes como son los controles VBX.

Browses

Los browses son controles especializados en la edición de bases de datos, su funcionamiento es parecido al browse original de CA-Clipper pero sin embargo su implementación es distinta ya que a diferencia de este los browses de FiveWin no contienen a su vez objetos columnas, lo cual es una de las limitaciones que tiene FiveWin, y es el motivo por el cual con la ultima versión se ha introducido un nuevo browse con objetos columnas pero se ha mantenido el original.

Recomendamos utilizar el original de FiveWin siempre que le sea posible ya que es mucho más rápido.

La sintaxis es muy parecida a la del control List box con la diferencia de que se utiliza la cláusula FIELDS para indicar que se van editar campos de una base de datos:

@ <nRow>, <nCol> LISTBOX [ <oBrw> ] ;

                 FIELDS <cFields,...> ;

                 [ FIELDSIZES <aColSizes,...> ] ;

                 [ HEAD | HEADER | HEADERS <aHeaders,...> ] ;

                 [ ID <nId> ] ;

                 [ SIZE <nWidth>, <nHeigth> ] ;

                 [ ON CHANGE | ON CLICK <uChange> ] ;

                 [ ON RIGHT CLICK <uRClick> ] ;

                 [ ON LEFT DBLCLICK <uLDblClick> ] ;

                 [ OF | DIALOG <oDlg> ]

La cláusula HEADERS permite indicar el nombre de cabecera de cada una de las columnas.

La cláusula SIZE permite indicar el tamaño que tendrá el control, esta cláusula no es opcional su introducción ya que sino se calculará automáticamente.

La cláusula FIELDSIZES permite indicar el ancho de cada columna en pixeles, sino introduce esta cláusula el control se ajustará automáticamente, pero no siempre de forma correcta porque como Windows utilizar fonts o tipos de letra proporcionales un campo de 20 caracteres no siempre ocupa lo mismo en pantalla ya que la “i” por ejemplo ocupa bastante menos que la “M”, el control browse calcula el ancho de columna basándose en la longitud del campo multiplicado por el ancho de la letra “B”, es posible que cuando muestre campos todos en mayúsculas no puede observar en algún campo la descripción completa, por el contrario en campos muy largos con letras minúsculas le parecerá que se desaprovecha una gran cantidad de espacio. Una solución a esta problema sería evaluar todos los registros de la base de datos para ver cual es el más ancho pero esto llevaría una gran cantidad de tiempo, en cualquier caso, este problema no es tan grave ya que se puede modificar el ancho de cualquier columna en run-time simplemente haciendo un drag con el ratón.

Desde el comando se pueden manejar los eventos más típicos que suele tener un control Browse como son ‘ON CHANGE’  que es cuando se hace un click sobre el objeto browse y se varía el elemento activo y por lo tanto el puntero de la base de datos, ‘ON RIGHT CLICK’ cuando se pulsa el botón derecho del ratón o el ‘ON LEFT DBLCLICK’ que es cuando se realiza un doble-click con el botón izquierdo del ratón.

Es habitual sobre todo en entornos MDI el hacer que un browse ocupe todo la superficie de una ventana hija (Child), y que cuando redimensionemos dicha ventana se redimensione el control Browse, para ello FiveWin implementa un método en la clase tMdiChild,  que es el método :SetControl().

oChildWnd:SetControl(oBrowse)

El manejo de ventanas MDI será explicado en el capítulo de programación avanzada.

También podemos crear browses desde recursos sin embargo debido a que dicho control no es un control standard ni de Microsoft Windows ni del Resource Workshop, hay que indicar al WorkShop que queremos crear un control de usuario (custom control). En el próximo capítulo se mostrará como realizarlo desde el Resource Workshop, pero una vez realizado para implementarlo en nuestro programa haríamos:

REDEFINE LISTBOX [ <oBrw> ] ;

         FIELDS [ <cFields,...>] ;

       [ FIELDSIZES <aColSizes,...> ] ;

       [ HEAD | HEADER | HEADERS <aHeaders,...> ] ;

       [ ID <nId> ] ;

       [ ON CHANGE | ON CLICK <uChange> ] ;

       [ ON RIGHT CLICK <uRClick> ] ;

       [ ON DBLCLICK <uDblClick> ] ;

       [ OF | WINDOW | DIALOG <oDlg> ]

El browse tiene una serie de métodos específicos para bases de datos como es la posibilidad de establecer un filtro en la misma, la posibilidad de utilizar un array en vez de una base de datos, etc..

Un método muy utilizado de la browse es el método ‘lEditCol’ que permite editar en linea el campo en el que se encuentre  el cursor, por ejemplo sí desea editar el campo cuando se realice un doble-click haría:

oBrowse:blDblClick := {|nRow, nCol, nFlags| ;




       oBrowse:lEditCol(nCol,...)

Consulte la guía Norton para obtener mayor información sobre su uso.

Otro método muy utilizado de la clase Browse es el método :Report() ya que nos permite realizar un informe por impresora del browse, es realmente cómodo para pequeños informes.

Meters

El control Meter se utiliza para mostrar barras de progreso y su utilización puede ser muy variada desde un proceso de reindexación, cálculo complejo sobre una base de datos, etc.

FiveWin hace realmente sencillo el uso de barras de progreso siendo su sintaxis la siguiente:

@ <nRow>, <nCol> METER [ <oMeter> VAR ] <nActual> ;

                 [ TOTAL <nTotal> ] ;

                 [ SIZE <nWidth>, <nHeight> ];

                 [ OF <oWnd> ] ;

                 [ UPDATE ] ;

                 [ PIXEL ] ;

                 [ FONT <oFont> ] ;

                 [ PROMPT <cPrompt> ] ;

                 [ NOPERCENTAGE ] ;

                 [ COLOR | COLORS <nClrPane>, <nClrText> ] ;

                 [ BARCOLOR <nClrBar>, <nClrBText> ] 

La característica fundamental de un Meter es el total que va a representar y el número actual representado, es decir, el total en una reindexación serían el número de registros y el número actual sería el Recno, es decir,

local nRecno := Recno()  ,;

      nTotal := LastRec()

@ <nRow>, <nCol> METER oMeter VAR nRecno ;

                 TOTAL nTotal ;

                 OF oDlg

Cuando queramos actualizar el meter haremos los siguiente:

nRecno += 1

oMeter:Refresh()

ó

oMeter:Set(nRecno+1)

También podemos crear el Meter desde recursos, la sintaxis de ‘REDEFINE’ es la siguiente:

REDEFINE METER [ <oMeter> VAR ] <nActual> ;

               [ TOTAL <nTotal> ] ;

               [ ID <nId> ] ;

               [ OF <oWnd> ];

               [ UPDATE ] ;

               [ FONT <oFont> ] ;

               [ PROMPT <cPrompt> ] ;

               [ NOPERCENTAGE ] ;

               [ COLOR | COLORS <nClrPane>, <nClrText> ] ;

               [ BARCOLOR <nClrBar>, <nClrBText> ] ;

Bitmap (Control)

El control bitmap tiene como misión el manejo de imágenes en formato “bitmap”, este control nos permite colocar en una caja de diálogo cualquier imagen bitmap y de cualquier resolución y número de colores.

La imagen puede ser recogida bien de un archivo o bien de recursos.

El control Bitmap nos permite ajustar el bitmap a las dimensiones que deseemos utilizando el método Stretch o la cláusula Adjust en el comando, de la misma forma sí el bitmap es más grande que nuestro control podemos incluirle barras de desplazamiento simplemente añadiendo la cláusula Scroll. 

Su sintaxis es la siguiente:

@ <nRow>, <nCol> BITMAP [ <oBmp> ] ;

               [ NAME | RESNAME | RESOURCE <cResName> ] ;

               [ FILE | FILENAME | DISK <cBmpFile> ] ;

               [ NOBORDER | NO BORDER ] ;

               [ SIZE <nWidth>, <nHeight> ] ;

               [ OF | WINDOW | DIALOG <oWnd> ] ;

               [ ON CLICK | ON LEFT CLICK <uLClick> ] ;

               [ ON RIGHT CLICK <uRClick> ] ;

               [ SCROLL ] ;

               [ ADJUST ] ;

               [ CURSOR <oCursor> ] ;

               [ PIXEL ] ;

               [ MESSAGE <cMsg>   ] ;

               [ UPDATE ] ;

               [ WHEN <WhenFunc> ] 

También podemos crear el control desde recursos, la sintaxis de ‘REDEFINE’ es la siguiente:

REDEFINE BITMAP [ <oBmp> ] ;

                [ ID <nId> ] ;

                [ OF | WINDOW | DIALOG <oWnd> ] ;

                [ NAME | RESNAME | RESOURCE <cResName> ] ;

                [ FILE | FILENAME | DISK <cBmpFile> ] ;

                [ ON ClICK | ON LEFT CLICK <uLClick> ] ;

                [ ON RIGHT CLICK <uRClick> ] ;

                [ SCROLL ] ;

                [ ADJUST ] ;

                [ CURSOR <oCursor> ] ;

                [ MESSAGE <cMsg>   ] ;

                [ UPDATE ] ;

                [ WHEN <WhenFunc> ] 

Botón Bar (Button bar)

El botón Bar permite mostrar botones con una imagen en su interior dentro de las barras de botones que se implementan con la clase Button Bar que como recordará sirven para poner una barra de botones en una ventana para tener un acceso más rápido a las opciones de menú más usadas.

La imagen puede ser recogida desde un fichero bitmap o desde recursos.

Si incluye la cláusula GROUP se creara un separador entre el anterior botón definido y el actual, de la misma forma sí incluye la cláusula TOOLTIP “mensaje” se visualizará un mensaje en una ventana flotante justo debajo del botón cuando posicione el cursor encima del mismo.

Su sintaxis es la siguiente:

DEFINE BUTTON [<oBtn>] ;

              [ OF | BUTTONBAR <oBar> ] ;

              [ NAME | RESNAME | RESOURCE <cResName1>;

              [,<cResName2>] ] ;

              [ FILE | FILENAME | DISK <cBmpFile1>;

              [,<cBmpFile2>] ] ;

              [ ACTION | EXEC <uAction> ] ;

              [ GROUP ] ;

              [ MESSAGE <cMsg> ] ;

              [ ADJUST ] ;

              [ WHEN <uWhen> ] ;

              [ TOOLTIP <cToolTip> ]

También podemos crear el control desde recursos, la sintaxis de ‘REDEFINE’ es la siguiente:

REDEFINE BUTTON [<oBtn>] ;

     [ ID <nId> ] ;

     [ OF | BUTTONBAR <oBar> ] ;

     [ NAME | RESNAME | RESOURCE <cResName1> [,<cResName2>] ] ;

     [ FILE | FILENAME | DISK <cBmpFile1> [,<cBmpFile2>] ] ;

     [ ACTION | EXEC <uAction> ] ;

     [ MESSAGE <cMsg> ] ;

     [ ADJUST ] ;

     [ WHEN <uWhen> ] ;

     [ UPDATE <lUpdate> ] 

No tiene mucho sentido redefinir este control ya que no suele crearse desde recursos puesto que las barras de botones aunque pueden colocarse en diálogos no es habitual (sólo en ventanas), sin embargo este redefine se puede usar para colocar un bitmap en un botón convencional de Microsoft Windows que de otra forma no sería posible mas que usando los controles avanzados de Borland (Bwcc.dll) que se explican a continuación.

Uso de BWCC.DLL

En el Resource Workshop de Borland podemos observar que además de los controles standard de Microsoft Windows existen otros adicionales que se autodenominan ‘Borland ... style’ siendo los puntos suspensivos el equivalente control standard, es decir existe un ‘Borland Radiobutton Sytle’ por ejemplo, también hay otros controles que no tienen su correspondiente control standard como es el control Bitmap que nada tiene que ver con el de FiveWin pero que en cierta modo tienen la misma funcionalidad aunque el de FiveWin es más potente.

Su funcionamiento es exactamente igual al de su correspondiente control standard y la única precaución que tenemos que tener es la de cargar la librería Bwcc.dll cuando arranquemos la aplicación.

LoadLibrary( "BWCC.dll" )

Por supuesto dicha librería debe encontrarse en el directorio donde se ejecuto la aplicación o en los directorios \Windows o Windows\System.
Ejercicios

1. Realice un programa que arranque con una caja de diálogo con dos botones, uno para arrancar una segunda caja de diálogo y otro para abandonar la aplicación.

2. Realice un programa que arranque con una caja de diálogo que contenga un listbox con los nombres de todos los alumnos y que cuando se seleccione uno de ellos el contendido de otro control tipo Get se actualice con el alumno elegido, de la misma forma sí se edita el Get y se módifica el alumno que el cambio también se refleje en el listbox.

3. Realice un programa que contenga tres Gets y un botón de Aceptar, cuando pulse el botón se ha de comprobar que ninguno de los gets esta en blanco y sí alguno lo estuviese el programa deberá dar un pitido y dar focus al primer control que este en blanco, en caso contrario deberá cerrar el diálogo y mostrar un mensaje con los diálogos predefinidos bajos las funciones ‘Msg’ indicando que el proceso ha terminado correctamente.

4. Realice un programa que contenga un control Browse para editar la base de datos Alumnos.Dbf y además permita la edición on-line de cualquiera de sus campos.

RECURSOS DE WINDOWS

¿Que son los recursos?

Los Recursos son una de las partes más importantes en la programación bajo Microsoft Windows. Podríamos definir los recursos como el conjunto de datos que complementan a cualquier  aplicación para su correcto funcionamiento, es decir, cualquier programa consta de una lógica implementada a través de procesos y unos datos que nos permiten dar cuerpo a toda la aplicación. 

La parte de datos de cualquier aplicación puede ser almacenada completamente o en parte bien en una librería de enlace dinámico (DLL) o incluso dentro del ejecutable, dichos datos son también llamados recursos los cuales pueden ser manejados y modificados externamente fuera del programa.

La separación entre código y datos es lo que permite el comienzo de la “Programación visual” la cual permite diseñar todos los datos que va a manejar una aplicación visualmente para después incluirlos en nuestra aplicación final.
FiveWin esta diseñada para hacer uso extensivo del los recursos de Windows por lo que el uso de un editor de recursos es esencial en la programación con FiveWin aunque realmente no es absolutamente necesario (recuerde el comando REDEFINE).

Gracias a esta tecnología puede realizar auténticos programas ‘Data-Driven’, lo que significa que puede modificar los datos del programa sin necesidad de volver a compilar o relinkar la aplicación.

Lo más importante de este proceso es comprender como podemos crear un objeto desde un recurso y como definir o redefinir su comportamiento.

Existen tres métodos para crear un objeto con FiveWin:

1. @ nRow, nCol ...: Este es el método original de CA-Clipper el cual crea un objeto e inmediatamente lo muestra en pantalla, es por ejemplo el caso de @ nRow, nCol GET. En este caso es CA-Clipper el que crea el objeto y no esta basado en recursos.

2. DEFINE <ObjectClass> <ObjectName>: Este segundo método principal para crear objetos en FiveWin, pero sin embargo dicho objeto no es presentado para ello hay que activarlo posteriormente con el comando ACTIVATE... Este sistema tampoco esta basado en recursos.

3. REDEFINE .... ID ...: En este caso esta utilizando recursos, el objeto esta creado con anterioridad, pero usted necesita modificar o definir su comportamiento. Una vez que ha redefinido su comportamiento el objeto puede ser activado utilizando el comando ACTIVATE...

Si desea utilizar recursos en sus aplicaciones este es el proceso a realizar:

· Consiga un editor de recursos como Borland Resource Workshop. Este diseñador de recursos lo puede encontrar con casi todos los compiladores de Borland para Windows, como son Turbo C++, Borland C++ o Turbo Pascal.

· Empiece a diseñar cada una de las cajas de diálogo de su aplicación con el editor de recursos. Los recursos que diseñe pueden ser guardados dentro de una DLL o en un fichero RC (los ficheros RC son ficheros ASCII que pueden ser compilados y solapados a nuestro ejecutable). En el directorio \FiveWin\dll existe una DLL vacía que puede utilizar para guardar todos los recursos que quiera (haga una copia de la DLL vacía por cada proyecto que tenga). 

· Cuando diseñe sus cajas de diálogo deberá adjudicar un nombre distinto a cada una de ellas, dicho nombre le permitirá seleccionarlas desde su programa. Cada control que incluya en sus cajas de diálogo deberá tener un identificador único para dicho diálogo y será también a través de dicho identificador como podrá redefinir cada uno de los controles 

· Si ha creado los recursos en una DLL deberá indicar a su programa que la va utilizar con el siguiente comando:


SET RESOURCES TO <Fichero.DLL>

· Si ha creado los recursos en un fichero RC deberá compilarlo con un compilador de recursos (RC.EXE) e incluirlo en su ejecutable. Una vez creado el ejecutable realice lo siguiente:


RC -K Programa.Exe

¿RC o DLL?

Los dos sistemas son muy validos para guardar recursos, pero hay importantes diferencias entre ellos:

· La principal ventaja de usar fichero .RC es que puede utilizar DEFINES para reemplazar a los identificadores de los controles, lo que significa que no tiene necesidad de memorizar los identificadores de todos los controles existentes, es decir, puede crear un fichero .CH con los DEFINES necesarios y luego indicar al fichero .RC que también utilice dicho fichero.

· Otra ventaja es que no es necesario utilizar el comando: SET RESOURCES TO ...

· El único inconveniente del fichero .RC es que tendrá que compilar los recursos cada vez que genere el ejecutable, sin embargo este proceso es muy rápido.

· La principal ventaja de guardar recursos en una DLL es que puede realizar autenticas aplicaciones “Data-driven” y que una misma DLL puede ser compartida por varias aplicaciones.

· El principal inconveniente de las DLL es que no puede usar DEFINES para identificar a cada control.

Tipos de recursos

Los recursos son datos que definen la parte visible de un programa, por ejemplo cuando usted abre una caja de diálogo y hace un click en un botón para realizar una determinada tarea en su programa, usted esta utilizando realmente recursos. En adición a las cajas de diálogo y botones su programa puede incluir iconos, cursores, bitmaps (imágenes), menús y aceleradores de teclado.

Diálogos

Un diálogo es una ventana que comunica información al usuario y permite al usuario seleccionar entre varias opciones, como pueden ser abrir un fichero, elegir un color, buscar un texto, etc.

Menús

Los programas realizados bajo Microsoft Windows suelen llevar incorporados una barra de menú, cada elemento de menú puede a su vez incluir otro submenú. Por ejemplo casi todos los programas de Microsoft Windows llevan incorporado una opción de menú Archivo que incluye comandos para crear o abrir un fichero existente.

Le recomendamos realizar los menús directamente desde código ya que es mucho más rápido.

Aceleradores

Los aceleradores son combinaciones de teclado que el usuario presiona para realizar determinadas tareas. Por ejemplo la combinación de teclado “Alt+F4” es un acelerador para cerrar la ventana o diálogo activo.

En general los programadores bajo Microsoft Windows utilizan los aceleradores para duplicar funciones de menú, permitiendo al usuario 

acceder a dichas funciones de una forma más rápida.

Tablas de cadenas (String Tables)

Los string tables contienen texto como descripciones, preguntas, mensajes de error, etc. que se muestran en los programas. La principal ventaja de utilizar string tables es que puede traducir fácilmente luego la aplicación a otro idioma distinto por ejemplo.

Bitmaps

Un bitmap es una representación binaria de una imagen gráfica. Windows personalmente utiliza gran cantidad de bitmaps, por ejemplo son bitmaps las imágenes típicas de cualquier ventana como las barras de desplazamiento, el símbolo del menú de control o los botones de minimizar o maximizar. 

FiveWin incorpora la clase tBitmap para facilitar el manejo de los mismos, para introducir un bitmap lo más cómodo es utilizar el comando BITMAP:

@ <nRow>, <nCol> BITMAP [ <oBmp> ] ;

                [ NAME | RESNAME | RESOURCE <cResName> ] ;

                [ FILE | FILENAME | DISK <cBmpFile> ] ;

                [ NOBORDER | NO BORDER ] ;

                [ SIZE <nWidth>, <nHeight> ] ;

                [ OF | WINDOW | DIALOG <oWnd> ] ;

                [ ON CLICK | ON LEFT CLICK <uLClick> ] ;

                [ ON RIGHT CLICK <uRClick> ] ;

                [ SCROLL ] ;

                [ ADJUST ] ;

                [ CURSOR <oCursor> ] ;

                [ PIXEL ] ;

                [ MESSAGE <cMsg>   ] ;

                [ UPDATE ] ;

                [ WHEN <WhenFunc> ] 

Como puede observar en el comando la cláusula RESOURCE le permite indicar el nombre que tiene el bitmap en el fichero de recursos, y la cláusula FILE le permite indicar el nombre del fichero .BMP, lógicamente deberá utilizar una u otra cláusula.

Iconos

Los iconos son bitmaps pequeños, generalmente de 32x32 pixeles de tamaño, que se utilizan para representar a las aplicaciones cuando están minimizadas.

Toda ventana que pueda ser minimiza nos mostrará su icono sí lo tuviese incorporado en el ejecutable, para indicar que una ventana tiene un determinado icono deberá indicárselo cuando la defina.

FiveWin implementa un comando para crear el objeto tIcon y luego deberá utilizar al ya conocido comando DEFINE WINDOW para crear la ventana y asignarla el icono recién creado.

DEFINE ICON oIcon RESOURCE "Mi_Icono"

DEFINE WINDOW oWnd ICON oIcon

Si es el icono de la ventana principal deberá necesariamente haber creado dicho icono desde recursos y con un fichero .RC. en el resto de los casos el icono puede ser creado desde un fichero .ICO.

También podrá incluir un icono a modo de control en cualquier caja de diálogo bien desde recursos o creándolo directamente desde FiveWin.

Desde recursos:

REDEFINE ICON <oIcon> ;

        [ ID <nId> ] ;

        [ <resource: NAME, RESOURCE, RESNAME> <cResName> ] ;

        [ <file: FILE, FILENAME, DISK> <cIcoFile> ] ;

        [ ON CLICK <uClick> ] ;

        [ <of: OF, WINDOW, DIALOG> <oWnd> ] ;

        [ <update: UPDATE> ] ;

        [ WHEN <uWhen> ] ;

Desde código:

@ <nRow>, <nCol> ICON [ <oIcon> ] ;

        [ <resource: NAME, RESOURCE, RESNAME> <cResName> ] ;

        [ <file: FILE, FILENAME, DISK> <cIcoFile> ] ;

        [ <border:BORDER> ] ;

        [ ON CLICK <uClick> ] ;

        [ <of: OF, WINDOW, DIALOG> <oWnd> ] ;

        [ <update: UPDATE> ] ;

        [ WHEN <uWhen> ] 

Cursores

Un cursor es un pequeño bitmap de 32x32 pixeles de tamaño que representa la posición del ratón en la pantalla. Los programas bajo Microsoft Windows utilizan los cursores para indicar el tipo de tarea que se esta haciendo en cada momento. Ejemplo típicos de cursores son la flecha, el reloj de arena o el Hi-Beam de edición.

Toda ventana que pueda tener su propio cursor que se nos mostrará cuando coloquemos el cursor del ratón en su interior, tenga en cuenta que cuando nos referimos a ventana lo estamos haciendo en sentido amplio, es decir, un diálogo y un control también pueden tener su propio cursor ya que también son ventanas.

FiveWin implementa un comando para crear el objeto tCursor y luego deberá utilizar al ya conocido comando DEFINE WINDOW para crear la ventana y asignarla el cursor recién creado. De la misma manera casi todos los controles tiene la posibilidad de indicarle un objeto tCursor. 

DEFINE CURSOR oCurosr RESOURCE "Mi_Cursor"

DEFINE WINDOW oWnd CURSOR oCursor

Los cursores sólo pueden ser utilizados desde recursos.

Fonts

Los programas bajo Microsoft Windows utilizan fonts para definir el tipo de letra, el tamaño y estilo del texto a utilizar.

FiveWin utiliza la clase tFont para hacer más cómodo su uso. Para crear un font utilice el siguiente comando:

DEFINE FONT oFont NAME “ARIAL” SIZE 10,10

Puede especificar varias cláusulas cuando cree el font como son:

· NAME: El nombre del font, Arial, System, Times, etc..

· SIZE: El tamaño del font indicando ancho y alto. Recomendamos que siempre indique un 0 para el ancho y un número negativo para el alto, de esta manera Microsoft Windows se encarga de proporcionar el ancho al alto dado.

· BOLD: Si desea el font en negrita

· ITALIC: Si desea el font en itálicas

· UNDERLINE: Si desea el font subrayado

· FROM USER: Si utiliza esta cláusula la creación del font se hará de forma iterativa mostrándosele al usuario un caja de diálogo con todas las posibilidades de creación del font.

Toda ventana que pueda tener su propio font para todos los textos que muestre, tenga en cuenta que cuando nos referimos a ventana lo estamos haciendo en sentido amplio, es decir, un diálogo y un control también pueden tener su propio font ya que también son ventanas para ello puede especificar la cláusula FONT en la creación de los mismos.

DEFINE FONT oFont NAME “ARIAL” SIZE 0,-12

@ 2, 2 GET cText OF oWnd FONT oFont

Resource Workshop

En este apartado se de un breve recorrido sobre el Resource Workshop de Borland intentando únicamente que el alumno se encuentre confortable lo antes posible con esta potentísima herramienta y por lo tanto no tiene intención de mostrar todas y cada una de sus posibilidades, sí desea más información sobre este programa le recomendamos que lea en profundidad el libro que acompaña al Workshop, sobre todo el capítulo dedicado a la creación de cajas de diálogo.

Desde este editor de programas podremos crear todos los recursos anteriormente enunciados  a través de un entorno visual de muy fácil uso.

Como se comentó anteriormente los recursos pueden ser guardados en una DLL o bien un fichero RC, en el primer caso la descripción de cada uno de los recursos es guardado en el propio fichero DLL, sin embargo cuando se trata de ficheros RC, en ellos sólo se guarda la descripción de parte de los recursos que va a utilizar una aplicación pero sin embargo el contenido de ciertos recursos suelen estar guardados en ficheros aparte, así por ejemplo las imágenes son guardadas en ficheros .BMP, los iconos en ficheros .ICO y los cursores en ficheros .CUR.

Los programas bajo Windows utilicen números para identificar a cada uno de los recursos. Usted utiliza estos números en sus programas para acceder a cada uno de los recursos, sin embargo los números no son muy descriptivos y por ello se utilizan directivas ‘#DEFINE’ para asignar a cada número una definición simbólica. Normalmente bajo CA-Clipper se guardan las directivas DEFINE dentro de un fichero con extensión .CH.

Es importante que cada vez que cree un proyecto nuevo incluya en su fichero .RC la siguiente instrucción:

#include "winapi.ch"

Esta instrucción permite al compilador de recursos (RC.EXE) resolver todas las referencias a identificadores y los posibles estilos de cualquier recurso o control. En Winapi.ch se han incluido casi todos los posibles estilos de cada uno de los controles, pero es posible que alguno no lo esté, en dicho caso el compilador de recursos le avisará diciendo que no conoce un determinado identificador como por ejemplo el identificador BBS_BITMAP que identifica a un control como del tipo ‘Borland Bitmap Style’, cuando esto ocurra quiere decir que en el fichero Winapi.ch no hay ningún define que identifique a BBS_BITMAP con ningún número por lo que deberá incluir en el fichero .RC manualmente la definición de dicho identificador de la siguiente manera:

#define BBS_BITMAP       0x8000L

Para encontrar dicho número deberá buscarlo en el fichero de nombre Windows.h que se encuentra en el directorio \Include de su compilador de C que acompaña el Resource Workshop.

Desde el Resource WorkShop puede abrir un proyecto creado con un fichero .RC o a través de una DLL, pero también sí lo desea puede abrir simplemente un fichero bitmap, un icono o un cursor, simplemente para editarlo.

Le recomendamos que no utilice el Resource Workshop para crear sus menús y aceleradores ya que es mucho más cómodo crearlos directamente desde FiveWin.

Es muy fácil copiar recursos entre proyectos para ello deberá cargar Resource Workshop varias veces, abrir un proyecto distinto en cada uno de ellos y luego utilizar las opciones de Copiar y Pegar para ir traspasando recursos de un proyecto a otro.

La mejor forma de aprender a utilizar el editor de recursos es practicando y cuando tenga alguna duda consulte la ayuda On-line que acompaña al programa.

Cajas de Diálogo

Para crear una caja de diálogo solo deberá utilizar la opción nuevo recurso, diálogo y a continuación introducir todos los controles que su nueva caja va a tener. 

Para introducir controles deberá seleccionar primero el control que desee con el ratón dentro de la ventana de controles disponibles y a continuación hacer un click en el área de la caja de diálogo donde desea ubicar el control.

Una vez creado el diálogo podrá cambiar su tamaño o moverlo con el ratón. Para cambiar el estilo del dialogo realice un doble-click en cualquier parte de la caja de diálogo y se le mostrará la ventana de estilos de la caja de diálogo, no se asuste con la cantidad de estilos posibles ya que no necesita conocer más que varios de ellos por lo menos al principio y además la mayoría de ellos son muy intuitivos.

Para obtener información sobre cada uno de los estilos consulte la ayuda On-line.

Algunos de los estilos no son incluso manejados por FiveWin por lo que no tiene sentido utilizarlos.

Trabajando con controles

Los controles se incluyen dentro de las cajas de diálogo para pasar y acceder información a los usuarios. Si desea obtener más información respecto a los controles consulte el anterior capítulo de este Curso.

En el Resource Workshop existe una paleta de controles disponible siempre que editemos cajas de diálogo que permiten de una manera cómoda e intuitiva la inclusión de cualquier control en la caja de diálogo en las coordenadas y con las dimensiones que queramos.

Cada uno de los controles tiene a su vez estilos que pueden ser mostrados realizando un doble-click en cada uno de ellos.

Tab Stop

Una vez introducidos todos los controles en nuestra caja de diálogo podremos identificar cuales recibirán focus cuando se pulse el tabulador y cuales no, por defecto el Resource Workshop dependiendo del control dará el estilo ‘Tab Stop’ al control pero sin embargo puede cambiarlo sí lo desea, por ejemplo un texto fijo no tiene sentido que tenga este estilo.

Para establecer dicho estilo en cada control puede acceder a la ventana de estilo de cada uno de los controles o bien utilizar la herramienta que incorpora el Resource Workshop en la paleta de herramientas la cual simplifica tremendamente el trabajo.

Agrupando controles

La agrupación de controles es una técnica que permite a los usuarios moverse entre un grupo de controles con las flechas de cursor, es por ejemplo el caso de los radio buttons, cuando se establece una agrupación de controles sólo debe tener el estilo ‘Tab Stop’ el primer control del grupo.

Para definir un grupo de controles,

1. Si es necesario mueva los controles hasta que se encuentren todos juntos.

2. Haga click sobre la paleta de herramientas que muestra una “G” entonces el cursor se convertirá en una “G” y cada uno de los controles que es el primer miembro de un grupo se verá remarcado con un rectángulo.

3. Por cada grupo que quiera definir haga un click en el primer miembro de cada uno de ellos de tal forma que vea como es remarcado por un rectángulo.

No es necesario que identifique el último miembro de un grupo. 

Esta técnica de grupo se utiliza también cuando quiere proporcionar un acelerador a un control que no tiene título como puede ser un Get, en este caso se establece un grupo entre dicho Get y el Say que suele tener al lado, se le asigna un acelerador al Say y se le da la propiedad al Say de primer miembro de grupo, además el Get debe ser el siguiente control después del Say en el orden en que se accede.

Ordenación de controles

Con esta herramienta podemos modificar el orden en que se accede a los controles. El orden es especialmente importante cuando ha definido grupos con los controles, sí los controles no son ordenados perfectamente el usuario puede volverse muy confuso con la utilización de nuestros programas.

Para establecer el orden utilice la herramienta de ordenación que se encuentra en la paleta de herramientas representada por un botón con los número “1,2”, cuando pulse dicho botón el Resource Workshop le mostrará el número de orden de cada control, simplemente haga click sobre cada control para establecer el nuevo orden, al primer control sobre el cual haga el click se convertirá en el control número 1, los siguientes click que realice aumentaran sucesivamente el contador de ordenación hasta que haya realizado un click sobre todos los controles.

Herramientas de alineación y ajuste de tamaño

El Resource Workshop incluye unas herramientas para alineación o ajuste de tamaño de controles, esta herramienta es muy cómoda y ahorra un montón de trabajo ya que no permite por ejemplo forzar a que varios controles tenga la misma altura, el mismo ancho, estén separados vertical o longitudinalmente el mismo espacio, estén centrados vertical o longitudinalmente en la caja de diálogo, etc.

Durante el curso se mostrará como realizar todas estas operaciones.

Probando una caja de diálogo

No es necesario que compile y linque su programa para observar como funciona su recién creada caja de diálogo ya que el Resource Workshop proporciona una herramienta para testearla, simplemente pulse el botón ‘Test’ de la paleta de herramientas..

Controles creados por el usuario

Es posible indicar al Resource Workshop que vamos a utilizar controles que en principio no están disponibles en el editor de recursos para ello hay que pulsar el botón de Custom Control representando por una llave en la paleta de herramientas e indicar el nombre de clase que va a tener el nuevo control. Una vez introducido un custom control haga un doble-click para indicar el estilo del mismo, por ejemplo para incluir un control browse de FiveWin el nombre de clase sería ‘tWbrowse’  y el estilo:

0|WS_CHILD|WS_VISIBLE|WS_TABSTOP|WS_BORDER|WS_VSCROLL|WS_HSCROLL

Bitmaps, Iconos y cursores

La creación de bitmaps, iconos y cursores es realmente intuitiva y se parece mucho al PaintBrush que incorpora Windows por lo que creemos que no merece la pena extendernos en ello.

Ejercicios

1. Realice un programa que acceda a una caja de diálogo mediante un fichero .RC  y el mismo programa accediendo al diálogo a través de una librería DLL.

2. Realice un programa con una ventana principal que cuando se iconice muestre un icono creado en una librería DLL y cuando el cursor se encuentre dentro de la ventana se convierta en el signo “+”.

3. Realice un programa con una caja de diálogo con un control browse para acceder a la base de datos Alumnos.dbf.

4. Realice un programa con una caja de diálogo que contenga todos los tipos de controles disponible en el Resource Workshop a través de un fichero .RC.

IMPRESIÓN EN FIVEWIN

Impresión en Microsoft Windows

La principal diferencia existente en Microsoft Windows y MSDOS en lo que concierne a la impresión, es que bajo Microsoft Windows se maneja la impresora de forma gráfica. Ello significa que vamos a poder utilizar fonts de distintos tamaños, podremos realizar todo tipo de figuras geométricas como líneas, círculos, rectángulos, etc. e incluso podremos incluir bitmaps (imágenes) dentro de nuestros listados.

Fivewin automatiza todo el proceso ofreciendo el máximo de potencia y flexibilidad y la mínima complejidad posible. 

FiveWin implementa la impresión a través de un objeto tPrinter el cual hace que esta tarea sea realmente sencilla, además FiveWin incorpora técnicas para rápidamente sacar algunos informes sin necesidad de programar, como por ejemplo:

· Para generar un informe de la base de datos activa simplemente haga una llamada a la función Report().

· Si desea generar un informe de un browse ejecute el método :Report() de dicha clase.

· Si quiere realizar una impresión de cualquier control, diálogo o ventana ejecute el método :HardCopy(nZoom) del objeto.

Si desea realizar tareas algo más sofisticadas deberá bien utilizar directamente el objeto tPrinter o utilizar el generador de Informes que incluye FiveWin.

Para crear un objeto tPrinter simplemente ha de utilizar los siguientes comandos xBase:

PRINT oPrn NAME “Mi primer listado” PREVIEW


   PAGE

      ...

      oPrn:Say(nRow, nCol, cText, ...)

      ...

     ENDPAGE

ENDPRINT

El comando PRINT crea un objeto tPrinter, desde ese momento ya puede manipular la impresión que los métodos y datos que ofrece dicha clase.

Puede incluir la cláusula FROM USER la cual muestra una caja de diálogo con todas las impresoras disponibles para que el usuario elija la que desee, sino incluye esta cláusula la impresora seleccionada será la que tenga definida por defecto en el panel de control o sí ha utilizado la función PrinterSetup() la que ha definido.

La función PrinterSetup() nos permite elegir una impresora con todas sus opciones de resolución, dimensión de pagina, etc.. pero sólo para la aplicación actual, es decir, no se grabarán los cambios que realice, esta misma función es usada por multitud de programas como por ejemplo el Notepad en la opción “Especificar impresora”.

La cláusula PREVIEW permite ver en pantalla cada una de las páginas a imprimir, pudiendo hacer Zoom de cualquiera de ellas e imprimir las que se deseen.

Normalmente creará varias paginas con las secuencias

PAGE

...

...

ENDPAGE

Y dentro de las mismas hará llamadas al método Say de la clase tPrinter para imprimir el texto que desee, tenga en cuenta que los parámetros de fila y columna que recibe el método Say hay que darlos en pixeles y dependiendo de la resolución de la impresora los resultados serán distintos, por este motivo la clase tPrinter ofrece varios métodos para resolver este problema:

· :nVertSize() que nos devuelve la longitud vertical de la página en milímetros

· :nHorizSize() que nos devuelve la longitud horizontal de la página en milímetros

· :nVertRes() que nos devuelve la resolución vertical de la página en milímetros

· :nHorizRes() que nos devuelve la resolución horizontal de la página en milímetros

· :nLogPixelY() que nos devuelve las resolución por pulgada vertical (por ejemplo 300 p.p.p típico de las láser)

· :nLogPixelX() que nos devuelve las resolución por pulgada horizontal

· :Cmtr2Pix(nRow, nCol): que transforma coordenadas en centímetros a coordenadas en pixeles

· :Inch2Pix(nRow, nCol): que transforma coordenadas en pulgadas a coordenadas en pixeles

Realmente los dos que más va usar van a ser estos dos últimos.

El problema de las distintas resoluciones de cada impresora también nos va a afectar con los fonts ya que cuanto mayor sea la resolución de la impresora más pequeños se imprimirán, para ello FiveWin incluye en el comando de creación del font la cláusula ‘OF’.

PRINTER oPrinter

DEFINE FONT oFont NAME “ARIAL” SIZE 0,-10 OF oPrinter

Si después interroga  a los datos nHeight y nWidth del objeto tPrinter se dará cuenta que tiene valores muy superiores debido a que han sido redimensionados directamente.

Recuerde destruir el font una vez hay terminado la impresión con el método End()

oFont:End() O RELEASE FONT oFont

El generador de informes

El generador de informes de FiveWin permite realizar casi cualquier listado de una forma muy sencilla, entendiendo por listado cualquier informe basado en columnas de información.

El generador de informes hace un uso muy extensivo de clases sin embargo gracias a los comandos su uso se hace realmente sencillo. 

La sintaxis para crear cualquier listado es la siguiente:

REPORT ....


COLUMN ...


COLUMN ...


[ GROUP ... ]

ENDREPORT

ACTIVATE REPORT ...

Primeramente deberá suministrar toda la información general del listado como es el título, cabecera, pie, fonts que va a utilizar, etc.

REPORT [ <oReport> ] ;

  [ TITLE <bTitle, ...> [< LEFT | CENTER | CENTERED | RIGHT > ] ];

  [ HEADER <bHead, ...> [< LEFT | CENTER | CENTERED | RIGHT > ] ];

  [ FOOTER <bFoot, ...> [< LEFT | CENTER | CENTERED | RIGHT > ] ];

  [ FONT <oFont, ...> ]  ;

  [ PEN <oPen, ...> ]  ;

  [ < SUMMARY >  ] ;

  [ <  FILE |  FILENAME |  DISK >  <cRptFile> ] ;

  [ <  NAME |  RESNAME | RESOURCE >  <cResName> ] ;

  [ <  TO PRINTER >  ] ;

  [ <  PREVIEW >  ] ;

  [ TO FILE <toFile> ] ;

  [ TO DEVICE <oDevice> ] ;

  [ CAPTION <cName> ]

Con cláusula TITLE indicará el título del listado pero sí se fija en el comando se dará cuenta que se pueden incluir varios títulos separados por comas, lo cual permite que el título ocupe más de una línea, además puede incluir la cláusula LEFT, CENTER o RIGHT, por defecto el título se mostrará centrado.

La cláusula HEADER y FOOTER tienen el mismo funcionamiento que la cláusula TITLE y sirven para especificar la cabecera y el pie del listado.

Puede utilizar cualquier expresión que devuelva una cadena de caracteres para definir el titulo, la cabecera o el listado y cada vez que vaya a imprimirse será evaluada, de esta forma es posible indicar por ejemplo que el pie de pagina es oReport:nPage y cada vez que se imprima el pie de pagina oReport:nPage será evaluado y por lo tanto indicará perfectamente el número de pagina del listado. En definitiva el texto que usted introduzca como descriptor del título, cabecera o pie de pagina es convertido en code-blocks, para poder así evaluarlo cada vez que haya que imprimirlo.

La cláusula FONT permite indicar separados por comas todos los objetos Fonts que va a utilizar nuestro Report, no es necesario que incluya la cláusula OF oPrinter cuando cree el Font ya que el propio generador de informes se encarga de redimensionar los fonts dependiendo de la resolución de la impresora. Una vez indicados los fonts para indicar por ejemplo que una columna va a utilizar el primer font de la lista simplemente hay que hacer:

COLUMN DATA... TITLE... FONT 1

Recuerde que ha de destruir los fonts cuando termine el listado.

La cláusula PEN permite indicar separados por comas todos los objetos PEN que va a utilizar nuestro Report, un PEN es un objeto que nos permite identificar el grosor y color de una línea, circulo o rectángulo, es decir, es el equivalente a un lápiz. Como verá más adelante es posible indicar al generador de informes que nos imprima líneas verticales remarcando columnas. Para crear un PEN utilice el siguiente comando:

DEFINE PEN oPen WIDTH 2

Recuerde que ha de destruir los Pens cuando termine el listado.

Al igual que en el caso de los fonts sí desea que una columna tenga unas rayas remarcándola deberá utilizar la cláusula GRID indicando el número de Pen a utilizar

COLUMN DATA... TITLE... GRID PEN 1

La cláusula SUMMARY permite imprimir sólo un sumario del listado, es decir, solamente la información de los grupos. lógicamente esta cláusula no tiene sentido usarla sino hay ningún grupo creado.

La cláusula TO PRINTER permite mandar directamente el listado a la impresora, PREVIEW nos muestra primeramente el listado por pantalla y TO FILE nos exporta el listado a un fichero ASCII. Si ya tiene creado el objeto tPrinter sobre el cual se va a realizar el listado puede indicárselo utilizando la cláusula TO DEVICE oPrinter.

La cláusula CAPTION permite indicar el título que se nos mostrará del listado en la ventana de previsualización y desde el administrador de impresión (Spooler).

El siguiente paso será indicar las columnas de dicho listado para lo cual disponemos de otro comando:

COLUMN [ <oRptCol> ] ;

       [ TITLE <bTitle, ...> ] ;

       [ AT <nCol> ] ;

       [ DATA <bData, ...> ] ;

       [ SIZE <nSize> ] ;

       [ PICTURE <cPicture, ...> ] ;

       [ FONT <uFont> ] ;

       [ <  TOTAL >  [ FOR <bTotalExpr> ] ] ;

       [ < LEFT | CENTER | CENTERED | RIGHT >  ] ;

       [ < SHADOW >  ] ;

       [ < GRID >  [ <nPen> ] ]

La cláusula TITLE nos permite indicar el título de la columna, el cual puede ocupar varias líneas para ello pondremos varios títulos separados por comas.

La cláusula DATA se utiliza para indicar el dato o los datos a mostrar, no es necesario que los convierta a cadena y puede utilizar cualquier expresión. Al igual que el título puede ocupar varias líneas.

La cláusula SIZE permite indicar el ancho de la columna, este ancho no es en pixeles sino en caracteres, no es necesario introducir este valor ya que el generador de informes lo calcula por usted.

La cláusula PICTURE permite establecer una máscara a utilizar para los datos, es muy importante que utilice la cláusula Picture para los campos numéricos que quiera totalizar en vez de hacer un Transform() del dato.

La cláusula FONT nos permite indicar que número de font se he de utilizar para la columna de todos los indicados cuando se creo el Report, aquí sí lo desea puede dar una expresión que devuelva un número en vez del número directamente, por ejemplo se podría hacer los siguiente:

COLUMN TITLE... DATA... FONT iif(Salario>100000,1,2)

de esta forma el dato se imprimiría con el font 1 cuando el salario fuera menor o igual a 100.000 y con el font 2 cuando fuera superior.

La cláusula TOTAL le permite especificar sí desea totales de dicha columna y además es posible incluir un condición de totalización, por ejemplo:

COLUMN TITLE... DATA... TOTAL FOR NOMBRE=“LUIS”

La cláusula LEFT|CENTER|RIGHT le permite indicar como se visualizará el texto tanto del título como de los datos.

La cláusula SHADOW hace que se muestre el fondo de la columna en un color gris claro y la cláusula GRID permite colocar dos líneas remarcando la columna.

Si lo desea puede establecer grupos de ruptura sobre todo para conseguir parciales de todos las columnas que haya totalizado, para ello deberá utilizar el siguiente comando:

GROUP [ <oRptGrp> ] ;

      [ ON <bGroup> ] ;

      [ HEADER <bHead> ] ;

      [ FOOTER <bFoot> ] ;

      [ FONT <uFont> ] ;

      [ < EJECT >  ]

Recuerde que puede incluir tantos grupos como desee.

GROUP ON Empleado->Edad

La cláusula ON le permite indicar la expresión de ruptura, es decir, sí queremos una ruptura cuando cambie la edad la expresión seria:

GROUP ON Empleado->Edad

Las cláusulas HEADER  y FOOTER nos permiten indicar una cabecera y un pie por cada grupo (sólo una línea).

La cláusula FONT funciona exactamente igual que en las columnas, con ellas indicamos el font a utilizar en la impresión de la ruptura.

La cláusula EJECT permite realizar un salto de página cada vez que cambia el grupo.

Para comenzar la impresión se debe utilizar el comando ACTIVATE REPORT al cual se le pueden incluir una serie de cláusulas que nos permiten especificar condiciones de impresión y modificar el listado mientras se esta generando.

Es en las  cláusulas ON donde se encuentra la verdadera potencia del generador de informes y cuando se tenga experiencia en el uso de ellas podrá realizar absolutamente cualquier informe.

La mejor forma de aprender a utilizar el generador de informes es ejecutar y analizar en profundidad el tutorial que acompaña a FiveWin.

El generador de informes utiliza 4 clases para realizar su cometido:

· Clase tReport

· Clase tRLine

· Clase tRColumn

· Clase tRGroup

La clase tReport contiene toda la información general de listado y es contenedor del resto de las clases.

La clase tRLine se utiliza par todas aquellas impresiones que se hacen longitudinalmente en todo el cuerpo del listado, es por ejemplo el título, la cabecera o el pie de página.

La clase tRColumn almacena toda la información de cada una de las columnas del listado. La clase tReport tiene un dato de nombre ::aColumns en el cual se guardan todos los objetos tRColumn.

Por último la clase tRGroup englobe todas las definiciones de grupo que hubiese en el listado.

Ejercicios

1. Realice un programa que contenga un browse con el fichero Alumnos.dbf y un botón para imprimir la base de datos utilizando el método que la clase browse incorpora .

2. Realice un programa que imprime una ficha completa de un alumno manejando directamente la clase tPrinter.

3. Genere un listado con el generador de informes de todos los campos del fichero Alumnos.dbf. ordenándolo por nombre y estableciendo un grupo en el campo sexo.

4. Intente incluir a dicho listado un bitmap en la esquina superior izquierda.

MANEJO DE BASES DE DATOS

Cambios en su programación

FiveWin no impone ningún tipo de restricciones o cambio en el diseño habitual de bases de datos desde CA-Clipper y de hecho puede continuar trabajando de la misma forma que lo hacia en MSDOS. Sin embargo hay ciertos puntos que ha de tener en cuenta:

· Por defecto Microsoft Windows sólo permite un máximo de 20 ficheros abiertos por aplicación, para conseguir abrir más ficheros deberá utilizar la función SetHanldeCount() que le suministra FiveWin la cual recibe como único parámetro el número de ficheros que desea abrir y devuelve los que realmente Microsoft Windows a podido ofrecerle. El número máximo de ficheros a abrir es 255, pero le recomendamos que indique los que realmente necesita su aplicación.

· Si desea mostrar una base de datos en dos ventanas distintas no tendrá más remedio que abrirla dos veces con distinto ‘Alias’, esto además le obliga a programar como sí se encontrará en un entorno de red, es decir, no podrá abrir las bases de datos en modo exclusivo y deberá bloquear los registros cada vez que realice una operación sobre ellos. Si esta utilizando Windows 3.1 necesitará cargar el programa SHARE.EXE antes de cargar Windows, para Windows 3.11 (trabajo en grupo) o Windows 95 no es necesario.

· FiveWin incorpora una potentísima clase DataBase la cual le puede ahorrar mucho tiempo de programación

· Si esta trabajando con versiones de CA-Clipper anteriores a la versión 5.2d puede tener serios problemas con algunos RDD sobre todo con el driver DbfCdx. En cualquier caso recomendamos que utilice el driver Comix o Six los cuales son excelentes.

El objeto Database

el objeto database ha sido diseñado con el único propósito de hacer más fácil el manejo de los ficheros DBF reduciendo considerablemente la cantidad de código necesario para su manejo. 

El objeto DataBase no intenta reemplazar completamente la forma en que accede a los ficheros DBF ya que muchas veces es incluso aconsejable su uso.

Un objeto DataBase es esencialmente un buffer en donde usted manipula los datos del fichero y que posteriormente graba en el fichero físicamente.

Para crearlo sólo necesita utilizar el comando DATABASE después de abrir la base de datos.

Local oDbf

USE “Alumnos” 

DATABASE oDbf

El objeto oDBF tendrá como datos todos los campos de dicho fichero, es decir, sí la base de datos tuviese un campo de nombre CODIGO existirá también en el fichero DBF un campo con el mismo nombre.

Cuando quiere que el objeto database cargue nuevos valores deberá utilizar el método :Load(), luego modifique sí lo desea cualquier campo utilizando los datos del objeto y luego grabe los cambios con el método :Save()

Tradicionalmente cualquier programador de CA-Clipper necesitaba crear variables locales que actuaran de buffer para almacenar cada uno de los valores de cada campo sobre todo cuando se programaba en entornos de red, otra alternativa era utilizar una matriz como buffer y utilizaban funciones para cargar y grabar, cuyo nombre más usado debido a que así lo implementa FoxPro son Scatter() y Gather(). El inconveniente de usar matrices es que no puede usar los nombres de campos directamente y por lo tanto hay que realizar DEFINES indicando que campo es cada elemento del array.

#define CODIGO
aDbf[1]

#define NOMBRE
aDbf[2]

El objeto database resuelve completamente este problema, ya que no necesitará crear más variables locales por cada campo, sólo necesitará crear una única variable local para guardar el objeto database.

Desgraciadamente el objeto database tiene una limitación y es que no puede manejar nombres de campo de más de 9 caracteres, ya que internamente CA-Clipper coloca un guión bajo delante de cualquier dato de clase y estos no pueden tener más de 10 caracteres de longitud. En cualquier caso no es una limitación muy importante.

Ejercicios

1. Realice un programa que contenga un browse con el fichero Alumnos.dbf y un botón para editar el registro actual utilizando un objeto database, cuando termine la edición el browse deberá reflejar los cambios efectuados.

2. Modifique el programa anterior para que funcione en red de área local.

3. Realice un programa que abra dos ventanas a la vez una con el browse y otra con los datos del registro actual y que cuando mueva el browse automáticamente se actualice la ventana de registro. (debe utilizar un objeto database para la ventana de registro.

EL DEBUGGER DE FIVEWIN

Utilización del debugger

FiveWin ofrece su propio debugger o depurador, para usarlo sólo hay que incluir el flag /b cuando compile sus programas y cuando ejecute su aplicación el debugger aparecerá automáticamente en su pantalla.

Para que el debugger requiere una DLL para su correcto funcionamiento de nombre Fwdbf.dll.

Es mucho más difícil depurar una aplicación bajo Microsoft Windows que bajo MSDOS, ya que es imposible parar la actividad  que Windows realiza para interactuar con el usuario, por ejemplo cada vez que movemos el ratón Microsoft Windows manda mensajes a nuestra aplicación indicándolo y nuestra aplicación debe procesarlo.

En cualquier caso el depurador de FiveWin no es de demasiada utilidad ya que todavía le falta mucha operatividad, por eso recomendamos que se utilice mejor la función MsgInfo() en las partes de su programa donde desea un “Watch-point”, esta solución es muy valida ya que el proceso compilar-linkar de FiveWin con Blinker es muy rápido.

Cada vez que se produce un error en FiveWin se genera un fichero de nombre Error.log en el cual se muestra una gran cantidad de información.

Ejercicios

1. Realice un programa con el debugger e inspeccione el objeto ventana cambiando su títulos y además añada un break-point cuando se cierre la ventana.

PROGRAMACIÓN AVANZADA

Refresco del sistema

Cuando usted realice procesos que supongan una gran cantidad de tiempo como puede ser un bucle de rastreo en una gran base de datos ha de tener en cuenta que Windows no podrá refrescar todo el sistema hasta que usted termine, por ejemplo sí hubiera un reloj en alguna pantalla éste se quedaría completamente parado hasta que terminará su bucle.

Para resolver este problema FiveWin incorpora una función “SysRefresh()” la cual podemos llamar tantas veces como queramos que nos permitirá refrescar el sistema cuando hagamos procesos largos:

DO WHILE !Eof()

  ....

  SysRefresh()

  ...

  SKIP

ENDDO

Manejo de ventanas MDI

Una ventana MDI es aquella que a su vez contiene otras ventanas en su interior, las ventanas hijas nunca pueden salir del marco de la ventana padre pero sí pueden ser movidas, redimensionadas,  iconizadas, destruidas y creadas.

Realmente una ventana MDI es como el entorno de Microsoft Windows pero para una sola aplicación.

Si tuviéramos que diseñar personalmente el entorno MDI sería realmente un trabajo enorme, sin embargo Microsoft Windows nos ofrece todo lo necesario para realizarlo.

Para entender como funciona una ventana MDI hay que saber que realmente existe una ventana invisible que es la ventana cliente, es decir, en un entorno MDI hay tres clases de ventanas:

· Ventana principal: tMdiFrame que es la ventana contenedor de la ventanas hijas y es la que por ejemplo contiene la barra de menú.

· Ventana cliente: tMdiClient que es la ventana que maneja realmente a las ventanas hijas, es una ventana invisible.

· Ventanas hijas: tMdiChild que son las ventas hijas de la ventana MDI.

Crear ventanas MDI es realmente una labor sencilla, lo realmente complicado es tener en cuenta que nos encontramos en un entorno no-modal puro y por lo tanto debemos manejarlo con mucha precaución, por ejemplo sí una ventana hija tiene un fichero de texto y otra un fichero DBF cuando pulsemos la opción imprimir que pueda existir en el menú principal nuestro programa ha de saber que ventana es la que esta activa cuando se pulsa la acción de menú y actuar en consecuencia, de la misma forma sí no hay ninguna ventana hija la opción de menú imprimir debería estar deshabilitada, en definitiva un entorno MDI supone más trabajo para el programador para controlar todas las posibles situaciones que se pueden producir un entorno no-modal. Por todo ello recomendamos que sólo se realicen entornos MDI cuando todas las ventanas hijas que se vayan a abrir sean del mismo tipo como puede ser el caso de un tratamiento de textos. Por supuesto en un entorno MDI puede arrancar una opción determinada con una caja de diálogo y no una ventana hija.

Le recomendamos que no empiece a utilizar ventanas MDI hasta que no se encuentre realmente cómodo con FiveWin.

FiveWin nos permite indicar un menú diferente por cada ventana hija que cree pero sinceramente no le recomendamos que lo haga ya que consume un montón de recursos de Windows y además dejan de funcionar los aceleradores de menú en cuanto haya alguno duplicado. Pero sí quiere hacer un buen diseño deberá crear un menú para cuando no haya ninguna ventana hija y otro menú para cuando sí lo haya, tenga en cuenta que cuando cambie de ventana hija el menú deberá actualizarse lo cual puede ser un gran trabajo.

La única diferencia a tener en cuenta cuando cree una ventana MDI es que deberá incluir la cláusula MDI cuando utilice el comando de creación de la ventana.

DEFINE WINDOW <oWnd> ;

     [ FROM <nTop>, <nLeft> TO <nBottom>, <nRight> ] ;

     [ TITLE <cTitle> ] ;

     [ STYLE <nStyle> ] ;

     [ MENU  <oMenu> ] ;

     [ BRUSH <oBrush> ] ;

     [ ICON  <oIcon> ] ;

     [ MDI ] ;

     [ COLOR | COLORS <nClrFore> [,<nClrBack>] ] ;

     [ VSCROLL | VERTICAL SCROLL ] ;

     [ HSCROLL | HORIZONTAL SCROLL ] ;

     [ MENUINFO <nMenuInfo> ] ;

     [ NONE | SINGLE ] ;

     [ OF <oParent> ]

Una cláusula que sólo poseen las ventanas MDI (frame) es MENUINFO. Normalmente todas las ventanas MDI suelen tener incorporado como opciones de menú la posibilidad de poner las ventanas en cascada, lado con lado, iconizarlas o elegir una de ellas desde menú, con esta cláusula indicaremos en que número de menú  principal queremos añadir las nuevas opciones. A medida que vaya abriendo o cerrando ventanas en la opción de menú que haya elegido se irán reflejando los cambios.

Para crear cada una de las ventanas hijas sólo hay que añadir la cláusula MDICHILD e indicar cual es su ventana contenedor con la instrucción OF:

DEFINE WINDOW [<oWnd>] ;

      [ MDICHILD ] ;

      [ FROM <nTop>, <nLeft> TO <nBottom>, <nRight> ] ;

      [ TITLE <cTitle> ] ;

      [ BRUSH <oBrush> ] ;

      [ CURSOR <oCursor> ] ;

      [ MENU <oMenu> ] ;

      [ ICON <oIco> ] ;

      [ OF <oParent> ] ;

      [ VSCROLL ] ;

      [ HSCROLL ] ;

      [ COLOR | COLORS  <nClrFore> [,<nClrBack>] ] ;

      [ PIXEL ] 

Las ventanas hijas no pueden tener un menú propio pero sí pueden tener una barra de botones.

Normalmente querrá que la ventana hija tenga un browse o un get multi-line y que además cuando redimensiona la ventana hija se redimensione el browse o el get, para ello existe un método en la clase tMdiChild que es :SetControl()

@ 0,0 GET oControl... MEMO OF oWndChild





o 

@ 0,0 LISTBOX oControl BROWSE ... OF oWndChild

oWndChild:SetControl( oControl )

Menús flotantes

Como su nombre indica un menú flotante es un menú que puede aparecer en cualquier lugar de una ventana.

Para crear un menú flotante sólo hay que indicar la cláusula POPUP de la siguiente forma

MENU oMenu POPUP

    MENUITEM ...

    MENUITEM ...

ENDMENU

ACTIVATE POPUP oMenu OF oDlg AT nRow, nCol

Normalmente utilizará menús popup cuando se pulse por ejemplo el botón derecho del ratón y por lo tanto las coordenadas nRow, nCol serían las coordenadas del ratón cuando se pulso el ratón derecho. 

Static oWnd

Function Main()

DEFINE WINDOW oWnd...

ACTIVATE WINDOW oWnd ON RIGHT CLICK PopuMenu(nRow,nCol)

....

Function PopupMenu()


local oMenu

  MENU oMenu POPUP

    MENUITEM ...

    MENUITEM ...

  ENDMENU

  ACTIVATE POPUP oMenu OF oWnd AT nRow, nCol

Return NIL

Sin embargo no funciona por que nRow y nCol tienen las coordenadas relativas a la ventana y sin embargo el menú popup utiliza las coordenadas de toda la pantalla (Windows desktop) por lo que hay que hacer una pequeña conversión de coordenadas de ventana a coordenadas de pantalla:

local aCoors := {nRow, nCol}

...

ClientToScreen(::oWnd:hWnd, aCoors)

...

ACTIVATE POPUP oMenu OF oWnd AT aCoors[1], aCoors[2]

Los Brushes

Los brushes son objetos gráficos que automáticamente rellenan la superficie de otro objeto.

Se pueden crear brushes de un determinado color, con un estilo que corresponda a una textura ya definida e incluso en base a un fichero bitmap.

Su uso consiste en primero definirlo y a continuación incluirlo en la definición de la ventana o diálogo.

DEFINE BRUSH <oBrush> ;

       [ STYLE  <Style> ] ;

       [ COLOR <nRGBColor> ] ;

       [ FILE | FILENAME | DISK <cBmpFile> ];

       [ RESOURCE | NAME | RESNAME <cBmpRes> ]

...

DEFINE WINDOW oWnd BRUSH oBrush

Existen gran variedad de estilos predefinidos con los cuales puede experimentar:

· Diagonal

· Horizontal

· Vertical

· Fdiagonal

· Bdiagonal

· Cross

· Diagcross

· Borland

· Tiled

· Bricks

El Clipboard

El clipboard o portapapeles es un buffer en donde se guarda información al cual pueden acceder todas las aplicaciones de Microsoft Windows .

FiveWin tiene una clase para acceder al Clipboard la cual hace muy fácil su uso. Una de las limitaciones de FiveWin es que sólo es capaz de manejar información de tipo texto.

El primer paso a realizar es definir el objeto clipboard y activarlo

DEFINE CLIPBOARD oClp OF oWnd

ACTIVATE CLIPBOARD oClp

Una vez definido podremos utilizar los métodos de dicha clase para realizar todas las operaciones necesarias, que son:

· oClp:Clear() para borrar el clipboard

· oClp:SetData() para mandar datos al clipboard

· oClp:GetData() para recuperar datos del clipboard

No olvide cerrar el clipboard una vez haya terminado de usarlo.

Ficheros INI

Los ficheros INI son ficheros de texto que Microsoft Windows utiliza como ficheros de configuración de forma extensiva y que además permite que cualquier aplicación cree sus propios ficheros INI o modifique los propios de Microsoft Windows.

Como seguramente sabrá Microsoft Windows utiliza sobre todo un fichero INI para guardar su configuración que es el fichero Win.Ini que se encuentra en el directorio de Windows.

Los ficheros INI están divididos en secciones y dentro de cada sección se incluyen unas claves seguidas por el signo igual y el dato a guardar para dicha clave.

// Fichero WIN.INI

[Intl]

sLanguage=esp

En este ejemplo la sección Internacional de nombre abreviado “Intl” tiene una clave que es “sLanguage” cuyo valor actual es “esp”, como se puede imaginar esta línea del fichero Win.Ini esta indicando el idioma que ha elegido en la configuración del panel de control para la representación de números y fechas.

Como puede observar todo el trabajo que se puede realizar con ficheros INI es cargar o grabar un determinado valor en una clave dentro de una sección especifica.

FiveWin tiene implementada una clase tIni para manejar dichos ficheros. Primeramente deberá abrir el fichero que desee, sí no indica ningún nombre es que quiere acceder directamente a Win.Ini a continuación hace todas las operaciones que desee  y por ultimo cierra el fichero.

INI oIni FILE "Fichero.ini"

    GET cTitulo ;

        SECTION "Principal" ;

        ENTRY "Titulo" ;

        OF oIni ;

        DEFAULT "Titulo de la ventana"

    SET SECTION "Principal" ;

        ENTRY "Titulo" ;

        OF oIni ;

        TO cTitulo

DEFAULT

Como puede observar en la cláusula GET puede indicar un valor por defecto sí lo desea. 

Otra característica fundamental es que el dato puede ser una cadena, un número, una fecha o un valor lógico, lo cual evita tener que hacer ninguna transformación en los datos.

Los ficheros INI pueden serle de gran utilidad para guardar multitud de parámetros de su programa.

Timers

Los timers son temporizadores que nos permiten ejecutar una instrucción cada cierto tiempo. No es aconsejable el uso indiscriminado de los mismos y las tareas que hagan deben ser muy rápidas y cortas.

Un timer va siempre asociado a una ventana, es decir, sí se destruye la ventana se destruye el timer.

Para crear un timer hay primero que definirlo y posteriormente activarlo, puede desactivarlo con el comando Deactivate o destruirlo con el comando Release

DEFINE TIMER <oTimer> ;

       [ INTERVAL <nInterval> ] ;

       [ ACTION <uAction> ] ;

       [ OF | WINDOW | DIALOG <oWnd> ]

...

ACTIVATE TIMER <oTimer>

...

DEACTIVATE TIMER <oTimer>

RELEASE TIMER <oTimer>

El intervalo de los timers se refiere siempre en milésimas de segundo, es decir, sí indica que el intervalo es 1000 se realizará la instrucción indicada en la cláusula ACTION cada segundo.

Multimedia

FiveWin incorpora algunas posibilidades multimedia como es la posibilidad de reproducir ficheros sonido y de vídeo.

Para reproducir un fichero de sonido de extensión WAV se utiliza la función SndPlaySound() con el nombre del fichero que se quiere reproducir.

Para reproducir un fichero de vídeo de extensión AVI FiveWin posee una clase que es tVideo con sus correspondientes comandos

@ 0,0 VIDEO oVideo FILE “Mivideo.avi” SIZE 200,200 OF oWnd

@ 2,0 BUTTON “Play” OF oWnd ACTION oVideo:Play(1)

Para reproducir sonidos y vídeo es necesario que el ordenador tenga capacidades multi-media.

Controles VBX

Los controles VBX son controles específicos de Visual Basic especializados en la realización de determinadas tareas.

Dichos controles han tenido mucho éxito ya que actualmente existen controles para realizar un gran número de tareas de una forma muy cómoda, por ejemplo Visual Basic tiene un control para visualizar bitmaps, otro para realizar círculos, etc. Este éxito ha propiciado para que casi la mayoría de los lenguajes de programación bajo Microsoft Windows hayan dado soporte para su utilización.

Sin embargo los días de los controles VBX están contados ya que será sustituidos por unos nuevos controles de nombre OCX que sólo funcionarán en sistemas de 32 bits y se basarán en la utilización de OLE de forma extensiva, en cualquier caso no parece que dicho día este muy cerca ya que siguen creándose nuevos controles VBX por parte de muchas compañías.

Hay tres niveles de controles VBX debido a que hay tres versiones de Visual Basic en el mercado, FiveWin y casi todo el resto de productos incluso Visual C++ de Microsoft sólo soportan el nivel primero, por lo que cuando compre un control VBX asegúrese que es de nivel 1. Casi todos los fabricantes de controles intentan ser de nivel 1 ya que así su mercado potencial de venta es mucho mayor.

Gran cantidad de los controles de Visual Basic ya están implementados directamente por FiveWin por lo que no necesitará usarlos.

FiveWin se apoya en el fichero BiVbx.DLL de Borland para implementar el acceso a los controles VBX, dicho fichero se encuentra en los últimos compiladores de Borland para Windows y no es suministrado con FiveWin.

La última versión del Resource Workshop de Borland permite también trabajar con controles VBX y para ello lleva incorporada una opción para añadir dichos controles. FiveWin permite usar controles VBX tanto desde código como desde recursos. 

Los controles VBX se almacenan en ficheros con la misma extensión y para que poder utilizarlos en nuestros programas es necesario entregar dichos ficheros al futuro usuario y grabarlos en su directorio \Windows\System.

Su funcionamiento es muy parecido al de cualquier otro control pero dependiendo de su funcionalidad es muy posible que tengan propiedades y eventos muy específicos que no existan en ningún otro control que haya visto antes.

Antes de trabajar con cualquier VBX es importante que tenga documentación del mismo ya que sino es muy probable que sea incapaz de hacerlo funcionar correctamente. 

FiveWin incorpora una pequeña utilidad para inspeccionar los controles VBX. Dicho programa se encuentra en código fuente por lo que deberá compilarlo y linkarlo, su nombre es VbxClass.prg y se encuentra en el directorio \fw\Samples.
Para crear un control VBX desde código:

@ <nRow>, <nCol> VBX [<oVbx>] ;

          [ OF <oWnd> ] ;

          [ SIZE <nWidth>, <nHeight> ] ;

          [ <file: FILE, FILENAME, DISK> <cVbxFile> ] ;

          [ CLASS <cVbxClass> ] ;

          [ ON <cClause1> <uAction1> ;

          [ ON <cClauseN> <uActionN> ] ] 

Como puede observar su sintaxis es muy parecida  a la de cualquier control pero con una serie de diferencias:

· Debe especificar el fichero VBX a usar.

· Debe especificar la clase del control, información que suele ser proporcionada por el fabricante del control. Puede coincidir con el nombre del fichero pero no tiene por que, sino conoce el nombre no incluya esta cláusula ya que es opcional, sí su control no arranca perfectamente es posible que tenga que indicar el nombre de clase.

· Las cláusulas ON permiten cualquier verbo, es decir, usted puede incluir en el comando la cláusula ‘ON LUNES’ pero posiblemente no sirva para nada. Ha de fijarse en los eventos que posee el control pues son los que tiene que incluir en la cláusula ON. Por ejemplo es muy probable que se nuevo control responda a la cláusula ‘ON CLICK’.

Algunos controles no funcionan bien dentro de una ventana y sólo funcionan en cajas de diálogo. Siempre que vaya a utilizar un control VBX en una caja de diálogo utilice siempre un editor de recursos porque le evitará un montón de problemas.

Para crear (redefinir) un control VBX desde recursos:

REDEFINE VBX [<oControl>] ;

        [ ID <nId> ] ;

        [ OF <oDlg> ] ;

        [ COLOR <nClrFore> [,<nClrBack>] ] ;

        [ ON <cClause1> <uAction1> ;

        [ ON <cClauseN> <uActionN> ] ] 

como puede observar la sintaxis es muy parecida a la de cualquier control.

ODBC

ODBC es el standard propuesto por Microsoft para acceder a bases de datos desde aplicaciones Windows. Piense en que ODBC es a Windows lo que RDD a CA-Clipper.

CA-Clipper RDD es una arquitectura abierta para utilizar multitud de drivers que accedan a bases de datos, ODBC es la arquitectura  abierta a utilizar en todas las aplicaciones Microsoft Windows.

De la misma manera que CA-Clipper tiene un motor interno para el manejo de los RDD, Microsoft Windows posee su propio motor cuyo nombre es Odbc.dll el cual permite que cualquier driver ODBC se comunique con él, haciendo que para el programador sólo exista un único driver, es decir, el mismo código independientemente del sistema de ficheros que se utilice e incluso aunque se acceda a los datos a través de servidores SQL.

ODBC impone sus propias reglas de utilización por lo que tendrá que ajustarse a las mismas, sí esta familiarizado con lenguajes SQL no le costará mucho pero sí sólo a programado con bases de datos navegacionales (Cobol, dBase, Paradox) el proceso de aprendizaje puede ser alto ya que deberá aprender SQL.

ODBC no es la panacea sólo hay que comparar la rapidez de cualquier driver RDD de CA-Clipper con el driver de manejo de ficheros DBF que proporciona Microsoft en Visual Basic o Acsess, realmente las diferencias son tremendas, no obstante esta claro que ODBC es el futuro y que con el tiempo los drivers para ODBC serán mejores y las máquinas más potentes.

FiveWin + ODBC permite hacer algo que hasta ahora era imposible que es acceder a servidores SQL bajo entorno cliente-servidor, ya que de la misma forma que su programa es capaz de acceder a datos locales con un driver ODBC para ficheros Dbase, sólo hay que cambiar el driver para indicar que ahora los datos se van a encontrar en un servidor UNIX con un motor SQL de Oracle por ejemplo.

Si sus desarrollos van a ser utilizados en empresas medianas o grandes le recomendamos que empiece ya a desarrollarlos bajo ODBC en cualquier otro caso utilice los RDD de CA-Clipper.

FiveWin utiliza una clase para acceder a cualquier base de datos vía ODBC que es la clase tDbOdbc a la cual se le han implementado dos comandos principalmente uno para crear una conexión ODBC y otro para establecer las consultas.

Para hacer la transición más sencilla desde ficheros Dbase a ODBC FiveWin ha implementado la clase ODBC de forma que sigamos utilizando el mismo esquema dBase para acceder a los datos.

Para utilizar ODBC desde cualquier aplicación Microsoft Windows debe asegurarse que todos los usuarios que van a utilizar la aplicación tengan cargado en su ordenador el driver ODBC  y que además este configurado correctamente. ODBC necesita saber a priori que bases de datos van a ser utilizadas vía ODBC y que drivers van a utilizar, por lo que abra que configurarlo en cada una de las máquinas.

Para establecer una conexión ODBC deberemos incluir en el fichero .Prg la instrucción 

#include “sql.ch”

y utilizaremos el siguiente comando:

DEFINE ODBC <oDbc> ;

      [ NAME <cName> ] ;

      [ USER <cUser> ] ;

      [ PASSWORD <cPassword> ] ;

      [ CONNECT <cConnect> ] ;

      [ FROM USER ]

oDbc es la variable que recoge el objeto recién creado.

La cláusula NAME  es el nombre de la base de datos a utilizar que corresponde con el nombre que se le dio en la configuración de OD BC y no con el nombre de ningún fichero DBF. Una única conexión ODBC puede acceder a multitud de ficheros DBF, sin embargo cuando configuramos ODBC debemos indicar que ficheros van a ser accedidos a través de un determinado NAME.

Las cláusulas User y Password se utilizan en algunas conexiones ODBC y no siempre son necesarias.

La cláusula CONNECT se utiliza en el caso de una conexión a un motor SQL y hay que introducir el ‘Data Source Name’ que puede ser por ejemplo “Oracle-Serv1”.

Como puede observar en el comando se nos permite elegir la conexión ODBC de forma interactiva.

Si hubiera habido algún error en la conexión puede comprobarlo con el dato :nRetCode del objeto oDbc que tendrá un valor distinto de cero.

Una vez establecida la conexión podremos realizar cualquier consulta con el comando

ODBC oDbc EXECUTE cCommand

El cCommand es una cadena con cualquier instrucción SQL que sea soportada por el driver que este utilizando, como por ejemplo:

ODBDC oDbc EXECUTE “SELECT * FROM CLIENTES”

Hemos podido comprobar que este comando no funciona al menos en la última versión de FiveWin y que se ha implementado una clase nueva de nombre OdbcStmt que es la que se encarga de hacer las conexiones  físicas y mantener la información de cada consulta que se haga al objeto oDbc. Este nuevo objeto se encarga de establecer la conexión física con el driver ODBC de la siguiente forma:

oStmt := OdbcStmt:New(oDbc)

de la misma forma habrá que comprobar el dato oSmt:nRetcode para ver sí la conexión ha sido correcta.

Una vez establecida la conexión se puede utilizar el método :Execute() para realizar cualquier consulta. En el caso de que haya realizado una instrucción ‘SELECT’ el objeto oStmt tendrá toda la información del Cursor, un cursor en SQL es un handle  a una tabla temporal creada a través de la instrucción SELECT, con la instrucción Fetch obtenemos en SQL todas las filas de la tabla temporal, para más información consulte cualquier libro sobre SQL.

Recomendamos que analice en profundidad el programa de ejemplo Odbc1.prg en el cual se muestra la actual forma de utilizar ODBC.

Manejo de DLL´s

Desde FiveWin puede acceder a cualquier librería de enlace dinámico para utilizarla en sus propios programas.

Para poder utilizar cualquier función que se encuentre en una DLL debe primeramente definirla con el siguiente comando

DLL [STATIC] FUNCTION <FunName>( ;

            [ <Param1> AS <Type> ] ;

            [ <ParamN> AS <Type> ] ) ;

            AS <return> [PASCAL] LIB <DllName>

este comando deberá incluirlo al principio o final de su .Prg y nunca dentro de una función o procedimiento. 

FunName es el nombre de la función, recuerde que no puede tener más de 10 caracteres, Paramx indica el nombre de los parámetros que va a recibir la función y  type indica el tipo de dichos parametros que puede ser:

· Byte, char (1 byte de longitud)

· Word, Bool, Handle, hWnd, hDC (2 bytes de longitud)

· Long, String, Lpstr, Ptr (4 bytes de longitud)

· Double (8 bytes de longitud)

Return indica el tipo de dato que devolverá la función, los valores pueden ser los mismo que el tipo.

La cláusula Lib indica el nombre de la librería a usar. 

Por último la cláusula Pascal indica sí la función a la que se accede lleva convención Pascal o no, normalmente todas las funciones del API de Windows llevan convención Pascal. (cuando en C una función se define tipo Pascal no distingue entre mayúsculas ni minúsculas y además no coloca el guión bajo al principio de la función).

Una de las mejoras introducidas en la ultima versión de FiveWin es la posibilidad de pasar punteros a dichas funciones para ello sólo hay que pasar el parámetro por referencia.

Si su aplicación hace uso extensivo de funciones que se encuentren en una DLL es preferible realizar un pequeño módulo en C para acceder a las mimas ya que la velocidad aumenta enormemente, en el capítulo de conexión con el API se mostrará la forma de hacerlo.

Dynamic Data Exchange (DDE)

DDE es un sistema de comunicación entre aplicaciones. Para poder establecer una comunicación entre dos aplicaciones es necesario que una de ellas sea DDE cliente y la otra DDE servidor o viceversa, el DDE cliente es el  que iniciara la conversación y pedirá al servidor que ejecute una serie de acciones. La sintaxis de dichas acciones o verbos dependerán del servidor DDE.

FiveWin puede actuar tanto como cliente como servidor DDE. 

Para poder establecer la comunicación con un servidor DDE es necesario conocer todos los verbos que admite dicho servidor y aunque parezca mentira muy pocos fabricantes de software dan información exhaustiva de ello, por ejemplo y a modo de ejemplo la documentación que acompaña al SDK de Window sólo ofrece información DDE del administrador de programas (Progman.exe).

Si desea utilizar desde FiveWin cualquier herramienta de generación de informes realizado por otro fabricante como CA-Ret, Crystal Reports, R&R, Report Smith, etc. no tendrá más remedio que hacerlo a través de un enlace DDE.

Tenga en cuenta que DDE tiene limitaciones y exige que ambas aplicaciones estén cargadas en memoria.

FiveWin incorpora una clase y un conjunto de comandos para realizar enlaces DDE. 

Los pasos a realizar en cualquier enlace DDE son los siguientes:

· Establecer la conexión con el servidor DDE

· Activar la conexión

· Realizar la comunicación con el método Execute()

· Finalizar la conexión.

Para establecer la conexión utilizamos el siguiente comando

DEFINE <dde:DDE,LINK> <oDde> ;

       [ SERVICE <cService> ] ;

       [ TOPIC <cTopic> ] ;

       [ ITEM <cItem> ] ;

       [ VALID  <uEnd> ] 

La cláusula Service indica la aplicación que va actuar como DDE server, ha de tener la precaución de que este cargada dicha aplicación utilizando la función WinExec()

#DEFINE SW_MINNOACTIVE     7

...

nStatus := WinExec(“Server.Exe”,SW_MINNOACTIVE)

IF nStatus < 32 .and. nStatus != 16

   MsgStop(“Error no se pudo arrancar Server.exe”)

   QUIT

ENDIF

Las cláusulas Topic e Item son opcionales y su uso depende de sí así lo requiere el servidor DDE.

La cláusula Valid se utiliza para realizar cualquier operación cuando finaliza la conexión DDE.

Manejo de ficheros de Ayuda

En este capítulo se dará información de como acceder a un fichero de ayuda para Microsoft Windows, en ningún caso se explicará como confeccionarlo.

Existen en el mercado muy buenos productos para diseñar ficheros de ayuda de los que se pueden destacar dos de ellos, el primero y más espectacular es Visual Help de WinWare Inc. (CIS: 70272,1656)   el cual tiene un entorno visual muy atractivo, y el segundo es Quick and dirty help de Phil Allen software ambos productos son shareware y pueden encontrarse en multitud de BBS y CD-Roms, existen productos mas profesionales pero creemos que para la mayoría de los programadores son excesivamente complejos y de una dificultad innecesaria.

FiveWin incorpora una serie de funciones para acceder a la ayuda de Microsoft Windows, al principio de su aplicación deberá designar cual es su fichero de ayuda con la instrucción

SET HELP TO “Miayuda.hlp”

Una vez realizado esta instrucción podrá utilizar una serie de funciones para invocarla, las más importantes son las siguientes:

· HelpIndex() que activa la ayuda mostrando el índice de la misma. Se suele invocar HelpIndex() desde una opción de menú de Ayuda(Contenido.

· HelpSearch(cTexto) que activa la ayuda y busca la cadena cTexto, sino introduce ningún texto el programa le pedirá que lo introduzca usted.

· HelpTopic(nTopic) que permite activar en un determinado topic o apartado.

FiveWin en su última versión ha mejorado substancialmente su sistema de ayuda, en la actualidad sólo es necesario establecer el fichero de ayuda a utilizar, del resto se encarga FiveWin.

El sistema funciona de la siguiente manera, cuando usted cree una ventana, un diálogo o un control podrá indicar sí lo desea un número para identificar el topic de su ayuda que quiere le corresponde (oWnd:nHelpId) y como podrá observar casi todos los controles y ventanas tienen la posibilidad de asignarlo directamente con su comando, cuando el usuario pulse F1 en cualquier parte de la aplicación FiveWin buscara el identificador de la ayuda en la ventana o control que tenga focus y sí valiese cero ira ascendiendo a su clase padre hasta encontrar un identificador distinto de cero, sí no encontrase ningún identificador distinto de cero le mostrará el índice de la ayuda.

Casi todos los programas que realizan ayudas para windows crean un fichero de cabecera con los defines de todos  los topics existentes cuya extensión suele ser .hhh, por lo que sólo tendrá que hacer un INCLUDE de dicho fichero en sus ficheros .PRG y cuando indique el identificador de ayuda no utilice el número sino el símbolo descriptor del mismo en el fichero .HHH que es el mismo que utilizo con la herramienta de generación de la ayuda.

Si desea incluir un botón de ayuda para un diálogo en la cláusula ACTION de dicho botón ponga sólo 

oDlg:HelpTopic()

También es posible asignar un identificador de ayuda a cada opción de menú.

Ejercicios

1. Realice un programa que abra cualquier base de datos con un browse bajo un entorno MDI con una opción de menú para imprimirla.

2. Realice una ventana que cuando pulse el botón derecho le muestre un menú Popup con las opciones Uno, dos y tres

3. Coloque un brush cualquiera en el programa realizado anteriormente

4. Realice una caja de dialogo con un get y dos botones uno para hacer un copy del dato del Get al Clipboard y otro que haga un “Paste”, es decir, un pegado del clipboard al Get.

5. Realice un programa que guarde en un fichero INI el Titulo de la ventana  y sus coordenadas al abandonar la aplicación.

6. Realice un ventana que muestre la hora actual utilizando un timer

7. Cree una caja de diálogo en la cual se utilice el control Switch.Vbx.

8. Realice un programa que cree un nuevo grupo en el administrador de programas de nombre Prueba y con un elemento que ejecute el Notepad con el fichero “Autoexec.bat”. 

OBJECTS: MOTOR OOPS DE FIVEWIN

FiveWin incorpora a CA-Clipper un motor para programación orientada a objetos de nombre OBJECTS.

Como ha podido observar hasta ahora hemos estado utilizando multitud de objetos nuevos que no existen normalmente en CA-Clipper y que lógicamente han sido creados por FiveWin.

La gran ventaja que ofrece FiveWin es que además usted puede crear sus propias clases, heredar de las ya existentes e incluso sí lo desea modificar las actuales ya que FiveWin le ofrece todos los fuentes de las clases que utiliza.

De momento no ha construido ninguna clase nueva ni ha modificado ningún dato o método de ninguna ya que no ha sido necesario, sin embargo sería una lastima el no aprovechar en beneficio propio las posibilidades que el OOPS nos ofrece. Cuando se encuentre cómodo con la programación orientada a objetos se preguntará como alguna vez ha podido programar sin ella.

Jerarquía de clases

En lo que concierne al manejo de ventanas FiveWin utiliza una clase base que es tWindow de la cual heredan el resto de las ventanas y  diálogos y una clase intermedia de nombre tControl de la cual hereden a su vez todos los controles. El resto de las clases casi ninguna tiene clase superior.

TDDE
Dynamic Data Exchange

Tfont
Font Class

Twindow
Window base Class

  (Tdialog
Dialog Boxes

  (Tbar
Button bar

  (TMdiFrame
MDI support

  (TMdiClient
MDI support

  (TMdiChild
MDI support

  (TMsgBar
Messages bar

  (Tcontrol
Base Class for controls

     (Tbitmap
Pictures in Bitmaps

     (Tbutton
PushButtons

     (TCheckBox
Logical checkboxes

     (TComboBox
Get + ListBox functionality

     (Tget
Typical Clipper GET

     (Ticon
Icons as controls

     (TListBox
ListBox selection list

     (Tmeter
Meter controls

     (TMultiGet
Gets for Memo editing

     (Tradio
Radio controls

     (TScrollBar
ScrollBars controls

     (TVbX
Visual Basic controls

     (TWBrowse
Browse controls

Tmenu
Menu Class

  (TMenuItem
Menu MenuItems

Tblock
FiveWin disk-storable codeblocks

TMCI
Media Control interface 

  (Tvideo
AVI Video controls

TRadMenu
Radio Controls Menu

Tprinter
Printer Class

Treport
Report engine

TrColumn
Report engine

TrGroup
Report engine

TrLine
Report engine

Ttimer
Timer Class

Acceso a los datos y métodos

Se puede acceder a cualquier dato o método de una clase siempre que no este protegido y realmente en FiveWin hay muy pocos datos que estén protegidos ya que esta construido con la intención de ofrecer la máxima flexibilidad posible.

Si necesita modificar un dato de una clase, puede hacerlo, sin embargo ha de saber muy bien lo que esta haciendo. FiveWin utiliza los comandos para ocultar precisamente la complejidad de algunas construcciones de objetos que reciben un montón de parámetros. Sin embargo hay datos que tendrá que modificar sin duda en cuanto empiece a programar con FiveWin, por ejemplo sí desea cambiar el valor de un control Get deberá utilizar el método cText() de dicha clase, puede sí lo desea cambiar el datos y luego utilizar el método Refresh() pero le recomendamos que antes de cambiar cualquier dato de una clase compruebe sí hay algún método para realizar dicha tarea ya que posiblemente cuando cambie el dato haya que hacer algo más que posiblemente usted no se haya percatado o desconozca. Por ejemplo la clase Listbox tiene un dato que son los items de dicho listbox que son guardados en una matriz, usted puede añadir elementos a la matriz sí lo desea pero posiblemente produzca un error de ejecución, observe como además la clase Listbox posee un método de nombre AddItem() que lógicamente deberá usar para añadir nuevos elementos.

Es importante que analice todas los datos y métodos de cada clase que quiera manipular, su arquitectura es muy sencilla y en poco tiempo  será capaz de sacar el máximo partido a FiveWin accediendo directamente a los métodos y datos de cada clase.

Cuando una clase es heredada de otra superior dicha clase tiene sus datos propios y además los de sus clases superiores, y lo mismo ocurrirá con los métodos. Cuando coincida un método de una clase con otro de su clase superior siempre tiene precedencia el método de la clase hija pero podemos hacer que desde el método de la clase hija se ejecute también el método de la clase padre como veremos más adelante, por ejemplo podemos crear un objeto ventana heredada de la clase tWindow que solo compruebe sí se pulsa la tecla F4 y en caso contrario que sea su clase superior la que procese la pulsación de teclado.

Creación de nuevas clases

Podrá crear todas las clases que desee pero sólo una por cada .PRG, para crear una nueva clase esta es la sintaxis que debe utilizar:

#include "Objects.ch" o #include “Fivewin.ch”

CLASS <NuevaClase> [ FROM <ClasePadre> ]

      DATA <NombreDato1>
      DATA <NombreDato2>

      ...

      METHOD <NombreMetodo1>( [ <params,...> ] ) ;

             [ CONSTRUCTOR | SETGET | VIRTUAL | BLOCK ]

      METHOD <NombreMetodo2>( [ <params,...> ] ) ;

             [ CONSTRUCTOR | SETGET | VIRTUAL | BLOCK ]

ENDCLASS

Nuevaclase es el nombre de la nueva clase que vamos a crear, su nombre no puede exceder de 10 caracteres.

ClasePadre es el nombre de la clase superior a la que pertenece sí es una clase heredada.

NombreDatox es el nombre de cada uno de los datos que va a tener nuestra clase, puede poner tantos nombres desee bien separandolos por comas o poniendo varias sentencias DATA.

NombreMetodox es el nombre de cada uno de los métodos que va a tener nuestra clase, puede poner tantos métodos desee poniendo varias sentencias METHOD.

Paramx es una lista de posibles parámetros que puede recibir el método.

Los métodos pueden recibir varias cláusulas:

· Constructor que indica que es un método para construir una instancia de la clase, es decir, el objeto, es el que debe utilizar para crear cada objeto. Se caracteriza porque devuelve una referencia de sí mismo.

· Virtual que es un método que no hace nada pero que es muy cómodo para herencia o cuando esta creando y probando un clase.

· Inline es un método que se define en la misma línea de definición.

Los métodos que no son en línea deben ser codificados a continuación de la definición de la clase. Para codificarlos se utiliza la siguiente sintaxis:

METHOD <NombreMetodo>( [<params,...>] ) CLASS <MiClase>

  ....

  ....

return

Dentro de cualquier método podemos acceder a cualquier dato de nuestra clase simplemente utilizando la variable Self, por ejemplo, para acceder a un dato de nombre nOrden haríamos: Self:Orden, por comodidad OBJECTS permite reemplazar la palabra “Self:” por “::” por lo que quedaría ::Orden.

Si desea acceder a un dato o método de su clase superior utilice la variable “Parent(NombrePadre)”, sí desea acceder a su clase padre principal puede utilizar “Super”, por ejemplo sí una clase tiene un método del mismo nombre que su padre principal para que cuando el termine de ejecutar su propio método se ejecute el de su padre haríamos:

METHOD KeyDown(nKey) CLASS MiVentana

   ...

   IF nKey == KEY_F4

      RETURN ...

   ENDIF

RETURN Super:KeyDown(nKey)

Ejercicios

1. Realice un programa que muestre en ventana constantemente la posición del cursor del ratón.

2. Cree una nueva clase de nombre rectángulo con los datos ancho y alto y con el método constructor New y el método área 

3. Cree un método para la clase Get de forma que invierta el sentido de sus letras, es decir, sí en el Get hay al valor “123” cuando ejecute el método :Invertir() pasará a valer “321”.

CONEXIÓN CON EL API DE WINDOWS

Como Conectar?

En principio FiveWin lleva incorporado una serie de funciones para conectar con el API que usted puede utilizar dentro de su programa, sin embargo no tiene implementadas todas pero como verá en seguida es muy fácil implementar la que necesite.

Conectar con el API de Windows desde su aplicación es una tarea muy sencilla, sólo hace falta un pequeño conocimiento de programación en C, pero no demasiado. 

Lo primero que ha de tener en cuenta es que CA-Clipper utiliza su propio sistema de Stack para las variables y que por lo tanto no puede pasar parámetros directamente desde CA-Clipper a una función en C y de la misma tampoco podrá dicha función en C devolver parámetros a CA-Clipper, para resolver este problema CA-Clipper proporciona lo que ellos llaman el “Extended system of CA-Clipper”, que consiste en funciones _par...() para recuperar desde C parámetros pasados desde CA-Clipper y funciones _ret...() para devolver valores desde C a CA-Clipper.

CA-Clipper utiliza símbolos PASCAL con un máximo de 10 caracteres, por lo que sí va utilizar identificadores de Windows con menos de 10 caracteres es posible que obtenga símbolos duplicados, este el motivo por el cual FiveWin ha renombrado todas las funciones del API de menos de 10 caracteres poniéndolas un guión bajo delante de ellas. Si su función tiene menos de 10 caracteres recuerde redifinirla con un guión bajo en el archivo Winten.h.

Para compilar su pequeña función en C puede utilizar cualquier compilador pero deberá utilizar el modulo largo de memoria e indicar al compilador que no linque, para ello se utilizan por ejemplo desde Borland C++ los flags: -c -ml

Todos sus programa en C deberán comenzar por los siguientes includes:

#include <Winten.h>

#include <Windows.h>

#include <Clipapi.h>

Como cabecera de funciones se utiliza la palabra ‘CLIPPER’

CLIPPER MiFuncion(params)

{

......

}

FiveWin incorpora todos los fuentes de conexión con el API en los que podrá observar lo sencillo que es.

Un programa en C

Por ejemplo la función que nos permite reproducir un fichero .wav sería de la siguiente forma:

#include <WinTen.h>

#include <Windows.h>

#include <MMSystem.h>

#include <ClipApi.h>

//--------------------------------------------------------//

CLIPPER SNDPLAYSOU() // ND( <cFichero>, <nModo> )--> <lExito>

{

   _retl( sndPlaySound( _parc( 1 ), _parni( 2 ) ) );

}

//--------------------------------------------------------//

Observe los detalles de nuestra función:

· Hemos incluido el include de MMSystem.h ya que es ahí donde esta la función SndPlaySound() prototipada

· Hemos utilizado la cadena CLIPPER al comienzo de la función.

· Solo hemos utilizado 10 caracteres para la definición de la función 

· Es una buena costumbre poner como comentario los posibles caracteres a partir del 10 además de los parámetros que va a manejar y el que va a retornar

· Como puede observar _parc( 1 ) es una función que recoge el primer parámetro pasado a la función que es de tipo carácter, y _parn(2) es el segundo parámetro que es de tipo numérico.

· La función devuelve un valor lógico a través de la función _retl()

Si desea obtener más información le recomendamos que analice todos los fuentes que proporciona FiveWin en el directorio \Fw\Source\Winapi y consulte la documentación que ofrece CA-Clipper sobre su “Extended System”. 

Ejercicios

1. Realice una función en C que devuelva el directorio de Windows.

Botón





List box





Menú de sistema





Rectangulo (box)





Say (Text)





Radio button





Combo box





Barra de desplazamiento





Barra de desplazamiento









